

Normalization By Evaluation of Types in $R\omega\mu$

Alex Hubers

Department of Computer Science

The University of Iowa

Iowa City, Iowa, USA

alexander-hubers@uiowa.edu

Conference'17, Washington, DC, USA

2025. ACM ISBN 978-x-xxxx-xxxx-x/YYYY/MM

<https://doi.org/10.1145/nnnnnnnn.nnnnnnnn>

split environments into kinding environments Δ , evidence environments Φ , and typing environments Γ .)

Abstract

Hubers and Morris [2023] introduce an expressive higher-order row calculus called $R\omega$, which relies heavily on implicit type reductions according to a directed type equivalence relation. The authors fail to describe any metatheory of this relation, and so it is unclear if type-equivalence is decidable or if types have normal forms. This report answers in the affirmative: we describe the normalization-by-evaluation (NbE) of types in $R\omega\mu$, a row calculus extending $R\omega$ with recursive types and a novel *row complement* operator. Types are normalized to β -short, η -long forms modulo a type equivalence relation. Because the type system of $R\omega\mu$ extends System $F\omega\mu$, much of the type reduction is isomorphic to reduction of terms in the STLC. Novel to this report are the reductions of row, record, and variant types.

1 Introduction

Hubers and Morris [2023] introduce an expressive higher-order row calculus called $R\omega$, which relies heavily on implicit type reductions according to a directed type equivalence relation. Despite this reliance, the authors only provide a proof of *semantic soundness* that well-typed terms inhabit the denotations of well-kinded types. The authors do not characterize the shape of types in normal form, nor prove that the denoted types are sound and completed with respect to their equivalence relation. Hubers et al. [2024] extends the $R\omega$ language to $R\omega\mu$, which is $R\omega$ with recursive types, term-level recursion, and a novel *row complement* operator. Despite this extension, the authors fail to describe any metatheory of the equivalence relation, and so it remains unsolved if type equivalence is decidable or if types have normal forms. We argue next that these results ought to be established.

1.1 The need for type normalization

Metatheory is difficult, particularly in the presence of conversion rules, which $R\omega$ and $R\omega\mu$ have. The rule below states that the term M can have its type converted from τ to v provided a proof that τ and v are equivalent. (For now, let us

$$(\text{T-CONV}) \frac{\Delta; \Phi; \Gamma \vdash M : \tau \quad \Delta \vdash \tau = v : \star}{\Delta; \Phi; \Gamma \vdash M : v}$$

Conversion rules complicate metatheory. To list a few reasons:

1. Decidability of type checking now rests upon the decidability of type conversion.
2. Relatedly, users of the surface language may be forced to write conversion rules by hand if type equivalence is not shown to be decidable.
3. Conversion rules block proofs of progress. Let M have type t , let pf be a proof that $t = u$, and consider the term $\text{conv } M \text{ pf}$; ideally, one would expect this to reduce to M (we've changed nothing semantically about the term). But this breaks type preservation, as $\text{conv } M \text{ pf}$ (at type u) has stepped to a term at type t .
4. Inversion of the typing judgment $\Delta; \Phi; \Gamma \vdash M : \tau$ —that is, induction over derivations—must consider the possibility that this derivation was constructed via conversion. But conversion from what type? Proofs by induction over derivations often thus get stuck.

1.2 Type computation in $R\omega\mu$

Let us demonstrate the significant role of type computation in $R\omega\mu$ types.

1.2.1 Reifying variants, reflecting records. The $R\omega\mu$ family of languages is quite expressive, with succinct and readable types. To some extent, this magic relies on implicit type application, mapping, and type equivalence. Let us demonstrate with the types of two terms that witness the duality of records and variants.

$$\begin{aligned} \text{reify} &: \forall z : R[\star], t : \star. \\ &(\Sigma z \rightarrow t) \rightarrow \Pi (z \rightarrow t) \\ \text{reflect} &: \forall z : R[\star], t : \star. \\ &\Pi (z \rightarrow t) \rightarrow \Sigma z \rightarrow t \end{aligned}$$

The term reify transforms a variant eliminator into a record of individual eliminators; the term reflect transforms a record of individual eliminators into a variant eliminator.

The syntax above is elegant, but arguably so because it hides some latent computation. In particular, what does $z \rightarrow t$ mean? The variable z is at kind $R[\star]$ and t at kind \star , so this is implicitly a map. Rewriting as such yields:

```
reify : ∀ z : R[★], t : ★.
  (Σ z → t) → Π ((λs. s → t) $ z)
reflect : ∀ z : R[★], t : ★.
  Π ((λs. s → t) $ z) → Σ z → t
```

The writing of the former rather than the latter is permitted because the corresponding types are equivalent modulo a type equivalence relation (Figure 2).

1.2.2 Deriving functoriality. We can simulate the deriving of functor typeclass instances: given a record of `fmap` instances at type Π (Functor z), I can give you a Functor instance for Σz .

```
type Functor : (★ → ★) → ★
type Functor = λf. ∀ a b. (a → b) → f a → f b
fmapS : ∀ z : R[★ → ★].
  Π (Functor z) → Functor (Σ z)
```

Pay close attention: what is the type of Functor z ? This is another implicit map. Let us write it as such and also expand the Functor type synonym:

```
fmapS : ∀ z : R[★ → ★].
  Π ((λf. ∀ a b.
    (a → b) → f a → f b) $ z) →
  (λf. ∀ a b. (a → b) → f a → f b) (Σ z)
```

which reduces further to:

```
fmapS : ∀ z : R[★ → ★].
  Π ((λf. ∀ a b.
    (a → b) → f a → f b) $ z) →
  ∀ a b. (a → b) → (Σ z) a → (Σ z) b
```

Intuitively, we suspect that $(\Sigma z) a$ means "the variant of type constructors z applied to the type variable a . Let's make this intent obvious. First, define a "left-mapping" helper `??` with kind $R[\star \rightarrow \star] \rightarrow \star \rightarrow R[\star]$ as so:

```
r ?? t = (λ f. f t) $ r
```

Now the type of `fmapS` is:

```
fmapS : ∀ z : R[★ → ★].
  Π ((λf. ∀ a b.
    (a → b) → f a → f b) $ z) →
  ∀ a b. (a → b) → Σ (z ?? a) → Σ (z ?? b)
```

And we have something resembling a normal form. Of course, the type is more interesting when applied to a real value for z . Suppose z is a functor for naturals, $\{ 'Z \triangleright \text{const Unit}, 'S \triangleright \lambda x. x \}$. Then a first pass yields:

```
fmapS { 'Z ▷ const Unit, 'S ▷ λx. x } :
  Π ((λf. ∀ a b. (a → b) → f a → f b)
    $ { 'Z ▷ const Unit, 'S ▷ λx. x }) →
  ∀ a b. (a → b) →
  Σ ({ 'Z ▷ const Unit, 'S ▷ λx. x } ?? a) →
  Σ ({ 'Z ▷ const Unit, 'S ▷ λx. x } ?? b)
```

How do we reduce from here? Regarding the first input, we suspect we would like a record of `fmap` instances for both the `'Z` and `'S` functors. We further intuit that the sub-term $\{ 'Z \triangleright \text{const Unit}, 'S \triangleright \lambda x. x \} ?? a$ really ought to mean "the row with `'Z` mapped to `Unit` and `'S` mapped to `a`". At this point I will perform multiple steps of computation simultaneously so as to retain the reader's attention.

```
fmapS { 'Z ▷ const Unit, 'S ▷ λx. x } :
  Π { 'Z ▷ ∀ a b. (a → b) → Unit → Unit,
    'S ▷ ∀ a b. (a → b) → a → b } →
  ∀ a b. (a → b) →
  Σ { 'Z ▷ Unit, 'S ▷ a } →
  Σ { 'Z ▷ Unit, 'S ▷ b }
```

The point we arrive at is that the elegance of some $R\omega$ and $R\omega\mu$ types are supplanted quite effectively by latent type equivalence. Further, as values are passed to type-operators, the shapes of the types incur forms of reduction beyond simple β -reduction. In our case, we must map type operators over rows; we next consider the reduction of row completions.

1.2.3 Desugaring Booleans. Consider a desugaring of booleans to Church encodings:

```
type BoolF = { 'T ▷ const Unit,
  'F ▷ const Unit,
  'If ▷ λx. Triple x x x }
type LamF = { 'Lam ▷ Id,
  'App ▷ λx. Pair x x,
  'Var ▷ const Nat }
desugar : ∀ y. BoolF ≤ y, LamF ≤ y \ BoolF ⇒
  Π (Functor (y \ BoolF)) →
  μ (Σ y) →
  μ (Σ (y \ BoolF))
```

We will ignore the already stated complications that arise from subexpressions such as `Functor (y \ BoolF)` and skip to the step in which we tell `desugar` what particular row y it operates over. Here we know it must have at least the `BoolF` and `LamF` constructors. Let's try something like the following AST, using `++` as pseudonotation for row concatenation.

```
type AST = BoolF ++ LamF ++
  { 'Lit ▷ const Int, 'Add ▷ λx. Pair x x }
desugar AST : BoolF ≤ AST, LamF ≤ (AST \ BoolF) ⇒
  Π (Functor (AST \ BoolF)) →
  μ (Σ y) → μ (Σ (AST \ BoolF))
```

When `desugar` is passed `AST` for z , the inherent computation in the complement operator is made more obvious. What should `AST \ BoolF` reduce to? Intuitively, we suspect the following to hold:

```
AST \ BoolF = { 'Lit ▷ const Int,
  'Add ▷ λx. Pair x x,
  'Lam ▷ Id,
  'App ▷ λx. Pair x x,
  'Var ▷ const Nat }
```

Type variables	$\alpha \in \mathcal{A}$	Labels	$\ell \in \mathcal{L}$
Kinds	$\kappa ::= \star \mid L \mid R[\kappa] \mid \kappa \rightarrow \kappa$		
Predicates	$\pi, \psi ::= \rho \lesssim \rho \mid \rho \odot \rho \sim \rho$		
Types	$\mathcal{T} \ni \phi, \tau, \rho, \xi ::= \alpha \mid T \mid \tau \rightarrow \tau \mid \pi \Rightarrow \tau$ $\mid \forall \alpha : \kappa. \tau \mid \lambda \alpha : \kappa. \tau \mid \tau \tau$ $\mid \{\xi_i \triangleright \tau_i\}_{i \in 0 \dots m} \mid \ell \mid \# \tau$ $\mid \phi \$ \rho \mid \rho \setminus \rho$		
Type constants	$T ::= \Pi^{(\kappa)} \mid \Sigma^{(\kappa)} \mid \mu$		

Figure 1. Syntax

But this computation must be realized, just as (analogously) λ -redexes are realized by β -reduction.

1.3 Contributions

In summary, this report offers the following as contributions:

1. A normalization procedure for the directed $R\omega$ and $R\omega\mu$ type equivalence relation;
2. the semantics of a novel *row complement* operator;
3. proofs of soundness and completeness of normalization with respect to type equivalence; and
4. a complete mechanization in Agda of $R\omega\mu$ and the claimed metatheoretic results.

2 The $R\omega\mu$ calculus

For reference, Figure 1 describes the syntax of kinds, predicates, and types in $R\omega\mu$.

Labels (i.e., record field and variant constructor names) live at the type level, and are classified by kind L . Rows of kind κ are classified by $R[\kappa]$. When possible, we use ϕ for type functions, ρ for row types, and ξ for label types. Singleton types $\# \tau$ are used to cast label-kinded types to types at kind \star . $\phi \$ \rho$ maps the type operator ϕ across a row ρ . In practice, we often leave the map operator implicit, using kind information to infer the presence of maps. We define a families of Π and Σ constructors, describing record and variants at various kinds; in practice, we can determine the kind annotation from context. μ builds isorecursive types. Row literals (or, synonymously, *simple rows*) are sequences of labeled types $\xi_i \triangleright \tau_i$. We write $0 \dots m$ to denote the set of naturals up to (but not including) m . We will frequently use ε to denote the empty row.

The type $\pi \Rightarrow \tau$ denotes a qualified type. In essence, the predicate π restricts the instantiation of the type variables in τ . Our predicates capture relationships among rows: $\rho_1 \lesssim \rho_2$ means that ρ_1 is *contained* in ρ_2 , and $\rho_1 \odot \rho_2 \sim \rho_3$ means that ρ_1 and ρ_2 can be *combined* to give ρ_3 .

Finally, $R\omega\mu$ introduces a novel *row complement* operator $\rho_2 \setminus \rho_1$, analogous to a set complement for rows. The complement $\rho_2 \setminus \rho_1$ intuitively means the row obtained by removing any label-type associations in ρ_1 from ρ_2 . In practice, the type $\rho_2 \setminus \rho_1$ is meaningful only when we know that $\rho_1 \lesssim \rho_2$, however constraining the formation of row complements to just this case introduces an unpleasant dependency between predicate evidence and types. In practice, it is easy enough to totally define the complement operator on all rows, even without the containment of one by the other.

3 Type Equivalence & Reduction

We define reduction on types $\tau \rightarrow_{\mathcal{T}} \tau'$ by directing the type equivalence judgment $\Delta \vdash \tau = \tau' : \kappa$ from left to right, defined in Figure 2. We omit conversion and closure rules.

3.1 Normal forms

The syntax of normal types is given in Figure 3. We carefully define the normal type syntax so that no type $\hat{\tau} \in \hat{\mathcal{T}}$ could reasonably reduce further to some other $\tau' \in \hat{\mathcal{T}}$. Hence we write $\tau \dashrightarrow \tau$ synonymously with $\tau \in \hat{\mathcal{T}}$ to indicate that τ is well-kinded and has no further reductions. We define a normalization function in Agda to materialize this sentiment later.

Type variables	$\alpha \in \mathcal{A}$	Labels	$\ell \in \mathcal{L}$
Ground Kinds	$\gamma ::= \star \mid L$		
Kinds	$\kappa ::= \gamma \mid \kappa \rightarrow \kappa \mid R[\kappa]$		
Row Literals	$\hat{\rho} \ni \hat{\rho} ::= \{\ell_i \triangleright \hat{\tau}_i\}_{i \in 0 \dots m}$		
Neutral Types	$n ::= \alpha \mid n \hat{\tau}$		
Normal Types	$\hat{\mathcal{T}} \ni \hat{\tau}, \hat{\phi} ::= n \mid \hat{\phi} \$ n \mid \hat{\rho} \mid \hat{\pi} \Rightarrow \hat{\tau}$ $\mid \forall \alpha : \kappa. \hat{\tau} \mid \lambda \alpha : \kappa. \hat{\tau}$ $\mid n \triangleright \hat{\tau} \mid \ell \mid \# \hat{\tau} \mid \hat{\tau} \setminus \hat{\tau}$ $\mid \Pi^{(\star)} \hat{\tau} \mid \Sigma^{(\star)} \hat{\tau}$		

$$\begin{array}{c}
\boxed{\Delta \vdash_{nf} \hat{\tau} : \kappa} \quad \boxed{\Delta \vdash_{ne} n : \kappa} \\
\\
(K_{nf-NE}) \frac{\Delta \vdash_{ne} n : \gamma}{\Delta \vdash_{nf} n : \gamma} \quad (K_{nf-\setminus}) \frac{\Delta \vdash_{nf} \hat{\tau}_1 : R[\kappa] \quad \hat{\tau}_1 \notin \hat{\rho} \text{ or } \hat{\tau}_2 \notin \hat{\rho}}{\Delta \vdash_{nf} \hat{\tau}_2 \setminus \hat{\tau}_1 : R[\kappa]} \\
\\
(K_{nf-\rightarrow}) \frac{\Delta \vdash_{ne} n : L \quad \Delta \vdash_{nf} \hat{\tau} : \kappa}{\Delta \vdash_{nf} n \triangleright \hat{\tau} : R[\kappa]}
\end{array}$$

Figure 3. Normal type forms

Normalization reduces applications and maps except when a variable blocks computation, which we represent as a *neutral type*. A neutral type is either a variable or a spine of applications with a variable in head position. We distinguish ground kinds γ from functional and row kinds, as neutral types may only be promoted to normal type at ground kind

$$\begin{array}{c}
\boxed{\Delta \vdash \tau = \tau : \kappa} \quad \boxed{\Delta \vdash \pi = \pi} \\
\\
\text{(E-}\beta\text{)} \frac{\Delta \vdash (\lambda \alpha : \kappa. \tau) v : \kappa'}{\Delta \vdash (\lambda \alpha : \kappa. \tau) v = \tau[v/\alpha] : \kappa'} \\
\\
\text{(E-LIFT}\Xi\text{)} \frac{\Delta \vdash \rho : R[\kappa \rightarrow \kappa'] \quad \Delta \vdash \tau : \kappa}{\Delta \vdash (\Xi^{(\kappa \rightarrow \kappa')} \rho) \tau = \Xi^{(\kappa')} (\rho^\$ \tau) : \kappa'} (\Xi \in \{\Pi, \Sigma\}) \\
\text{where } \rho^\$ \tau = (\lambda f. f \tau) \$ \rho \\
\\
\text{(E-}\backslash\text{)} \frac{\Delta \vdash \rho_i : R[\kappa]}{\Delta \vdash \rho_2 \backslash \rho_1 = \text{subtract } \rho_2 \rho_1 : R[\kappa]} \\
\\
\text{(E-MAP)} \frac{\Delta \vdash \phi : \kappa_1 \rightarrow \kappa_2 \quad \Delta \vdash \{\xi_i \triangleright \tau_i\}_{i \in 0 \dots n} : R[\kappa_1]}{\Delta \vdash \phi \$ \{\xi_i \triangleright \tau_i\}_{i \in 0 \dots n} = \{\xi_i \triangleright \phi \tau_i\}_{i \in 0 \dots n} : R[\kappa_2]} \\
\\
\text{(E-MAP}_{\text{id}}\text{)} \frac{\Delta \vdash \rho : R[\kappa]}{\Delta \vdash (\lambda \alpha. \alpha) \$ \rho = \rho : R[\kappa]} \\
\\
\text{(E-MAP}_{\circ}\text{)} \frac{\Delta \vdash \phi_1 : \kappa_2 \rightarrow \kappa_3 \quad \Delta \vdash \phi_2 : \kappa_1 \rightarrow \kappa_2 \quad \Delta \vdash \rho : R[\kappa_1]}{\Delta \vdash \phi_1 \$ (\phi_2 \$ \rho) = (\phi_1 \circ \phi_2) \$ \rho : \kappa_3} \\
\text{where } \phi_1 \circ \phi_2 = \lambda \alpha. \phi_1 (\phi_2 \alpha) \\
\\
\text{(E-MAP}\backslash\text{)} \frac{\Delta \vdash \phi : \kappa_1 \rightarrow \kappa_2 \quad \Delta \vdash \rho_i : R[\kappa_1]}{\Delta \vdash \phi \$ (\rho_2 \backslash \rho_1) = \phi \$ \rho_2 \backslash \phi \$ \rho_1 : \kappa_2} \\
\\
\text{(E-}\Xi\text{)} \frac{\Delta \vdash \rho : R[R[\kappa]]}{\Delta \vdash \Xi^{(R[\kappa])} \rho = \Xi^{(\kappa)} \$ \rho : R[\kappa]} (\Xi \in \{\Pi, \Sigma\}) \\
\\
\text{(E-}\eta\text{)} \frac{\Delta \vdash \phi : \kappa_1 \rightarrow \kappa_2}{\Delta \vdash \phi = \lambda \alpha : \kappa_1. \phi \alpha : \kappa_1 \rightarrow \kappa_2} \\
\\
\boxed{\text{subtract } \rho \rho} \\
\\
\text{subtract } \varepsilon \rho = \varepsilon \\
\text{subtract } \rho \varepsilon = \rho \\
\text{subtract } \{\ell \triangleright \tau, \rho\} \{\ell' \triangleright \tau', \rho'\} = \begin{cases} \text{subtract } \rho \rho' & \text{if } \ell = \ell' \text{ and } \tau = \tau' \\ \{\ell \triangleright \tau, \text{subtract } \rho \{\ell' \triangleright \tau', \rho'\}\} & \text{if } \ell < \ell' \\ \text{subtract } \{\ell \triangleright \tau, \rho\} \rho' & \text{if } \ell > \ell' \end{cases}
\end{array}$$

Figure 2. Type equivalence

(rule $(\kappa_{nf}\text{-NE})$): neutral types n at functional kind must η -expand to have an outer-most λ -binding (e.g., to $\lambda x. n x$), and neutral types at row kind are expanded to an inert map by the identity function (e.g., to $(\lambda x. x) \$ n$). Likewise, repeated maps are necessarily composed according to rule (E-MAP_{\circ}) : For example, $\phi_1 \$ (\phi_2 \$ n)$ normalizes by letting ϕ_1 and ϕ_2 compose into $((\phi_1 \circ \phi_2) \$ n)$. By consequence of η -expansion, records and variants need only be formed at kind \star . This means a type such as $\Pi(\ell \triangleright \lambda x. x)$ must reduce to $\lambda x. \Pi(\ell \triangleright x)$, η -expanding its binder over the Π . Nested applications of Π and Σ are also "pushed in" by rule $(\text{E-}\Xi)$. For example, the

type $\Pi \Sigma (\ell_1 \triangleright (\ell_2 \triangleright \tau))$ has Σ mapped over the outer row, reducing to $\Pi(\ell_1 \triangleright \Sigma(\ell_2 \triangleright \tau))$.

The syntax $n \triangleright \hat{\tau}$ separates singleton rows with variable labels from row literals $\hat{\rho}$ with literal labels; rule $(\kappa_{nf}\text{-}\triangleright)$ ensures that n is a well-kinded neutral label. A row is otherwise an inert map $\phi \$ n$ or the complement of two rows $\hat{\tau}_2 \backslash \hat{\tau}_1$. Observe that the complement of two row literals should compute according to rule $(\text{E-}\backslash)$; we thus require in the kinding of normal row complements $(\kappa_{nf}\text{-}\backslash)$ that one (or both) rows are not literal so that the computation is indeed inert. The remaining normal type syntax does not differ meaningfully from the type syntax; the remaining kinding rules for the judgments $\Delta \vdash_{nf} \hat{\tau} : \kappa$ and $\Delta \vdash_{ne} n : \kappa$ are as expected.

3.2 Metatheory

3.2.1 Canonicity of normal types. The normal type syntax is pleasantly partitioned by kind. Due to η -expansion of functional variables, arrow kinded types are canonically λ -bound. A normal type at kind $R[\kappa]$ is either an inert map $\hat{\phi}^\star n$, a variable-labeled row $(n \triangleright \hat{\tau})$, the complement of two rows $\hat{\tau}_2 \backslash \hat{\tau}_1$, or a row literal $\hat{\rho}$. The first three cases necessarily have neutral types (recall that at least one of the two rows in a complement is not a row literal). Hence rows in empty contexts are canonically literal. Likewise, the only types with label kind in empty contexts are label literals; recall that we disallowed the formation of Π and Σ at kind $R[L] \rightarrow L$, thereby disallowing non-literal labels such as $\Delta \not\vdash \Pi \varepsilon : L$ or $\Delta \not\vdash \Pi(\ell_1 \triangleright \ell_2) : L$.

Theorem 3.1 (Canonicity). *Let $\hat{\tau} \not\rightarrow \tau$.*

- If $\Delta \vdash_{nf} \hat{\tau} : (\kappa_1 \rightarrow \kappa_2)$ then $\hat{\tau} = \lambda \alpha : \kappa_1. \hat{v}$;
- if $\varepsilon \vdash_{nf} \hat{\tau} : R[\kappa]$ then $\hat{\tau} = \{\ell_i \triangleright \hat{\tau}_i\}_{i \in 0 \dots m}$.
- If $\varepsilon \vdash_{nf} \hat{\tau} : L$, then $\hat{\tau} = \ell$.

3.2.2 Normalization.

Theorem 3.2 (Normalization). *There exists a normalization function $\Downarrow : \mathcal{T} \rightarrow \hat{\mathcal{T}}$ that maps well-kinded types to well-kinded normal forms.*

\Downarrow is realized in Agda intrinsically as a function from derivations of $\Delta \vdash \tau : \kappa$ to derivations of $\Delta \vdash_{nf} \hat{\tau} : \kappa$. Conversely, we witness the inclusion $\hat{\mathcal{T}} \subseteq \mathcal{T}$ as an embedding $\Uparrow : \hat{\mathcal{T}} \rightarrow \mathcal{T}$, which casts derivations of $\Delta \vdash_{nf} \hat{\tau} : \kappa$ back to a derivation of $\Delta \vdash \tau : \kappa$; we omit this function and its use in the following claims, as it is effectively the identity function (modulo tags).

The following properties confirm that \Downarrow behaves as a normalization function ought to. The first property, *stability*, asserts that normal forms cannot be further normalized. Stability implies *idempotency* and *surjectivity*.

Theorem 3.3 (Properties of normalization).

- (Stability) for all $\hat{\tau} \in \hat{\mathcal{T}}$, $\Downarrow \hat{\tau} = \hat{\tau}$.

- (Idempotency) For all $\tau \in \mathcal{T}$, $\Downarrow (\Downarrow \tau) = \Downarrow \tau$.
- (Surjectivity) For all $\hat{\tau} \in \hat{\mathcal{T}}$, there exists $v \in \mathcal{T}$ such that $\hat{\tau} = \Downarrow v$.

We now show that \Downarrow indeed reduces faithfully according to the equivalence relation $\Delta \vdash \tau = \tau : \kappa$. Completeness of normalization states that equivalent types normalize to the same form.

Theorem 3.4 (Completeness). *For well-kinded $\tau, v \in \mathcal{T}$ at kind κ , If $\Delta \vdash \tau = v : \kappa$ then $\Downarrow \tau = \Downarrow v$.*

Soundness of normalization states that every type is equivalent to its normalization.

Theorem 3.5 (Soundness). *For well-kinded $\tau \in \mathcal{T}$ at kind κ , there exists a derivation that $\Delta \vdash \tau = \Downarrow \tau : \kappa$. Equivalently, if $\Downarrow \tau = \Downarrow v$, then $\Delta \vdash \tau = v : \kappa$.*

Soundness and completeness together imply, as desired, that $\tau \longrightarrow_{\mathcal{T}} \tau'$ iff $\Downarrow \tau = \Downarrow \tau'$.

3.2.3 Decidability of type conversion. Equivalence of normal types is syntactically decidable which, in conjunction with soundness and completeness, is sufficient to show that $\mathcal{R}\omega\mu$'s equivalence relation is decidable.

Theorem 3.6 (Decidability). *Given well-kinded $\tau, v \in \mathcal{T}$ at kind κ , the judgment $\Delta \vdash \tau = v : \kappa$ either (i) has a derivation or (ii) has no derivation.*

4 Normalization by Evaluation (NbE)

This section and those that follow give a closer examination into how the above metatheory was derived. In particular, we explain the *normalization of types by evaluation* (NbE) involved in deriving a normalization algorithm. We describe the standard components of NbE, but place emphasis on the novelty of normalizing rows and row operators.

Normalization by evaluation comes in a handful of different flavors. In our case, we seek to build a normalization function $\Downarrow : \mathcal{T} \rightarrow \hat{\mathcal{T}}$ by interpreting derivations in $\mathcal{T}_{\Delta}^{\kappa}$ (the set of judgments of the form $\Delta \vdash \tau : \kappa$) into a semantic domain capable of performing reductions semantically. We then *reify* objects in the semantic domain back to judgments in $\hat{\mathcal{T}}_{\Delta}^{\kappa}$ (the set of judgments of the form $\Delta \vdash_{nf} \tau : \kappa$). The mapping of syntax to a semantic domain is typically written as $\llbracket \cdot \rrbracket$ and called the *residualizing semantics*. For example, a judgment of the form $\Delta \vdash \phi : \star \rightarrow \star$ could be interpreted into a set-theoretic function, allowing applications to be interpreted into set-theoretic applications by that function. In our case, the syntax of the judgments $\Delta \vdash \tau : \kappa$, $\Delta \vdash_{nf} \tau : \kappa$, and $\Delta \vdash_{ne} \tau : \kappa$ are represented as Agda data types (where Env is a list of De Bruijn indexed type variables and Kind is the type of kinds):

```
data Type : Env → Kind → Set
data NormalType : Env → Kind → Set
data NeutralType : Env → Kind → Set
```

4.1 Residualizing semantics

We define our semantic domain in Agda recursively over the syntax of Kinds. Care must be taken to not run afoul of Agda's termination and positivity checking.

```
SemType : Env → Kind → Set
SemType Δ ★ = NormalType Δ ★
SemType Δ L = NormalType Δ L
SemType Δ1 (κ1 → κ2) = KripkeFunction Δ1 κ1 κ2
SemType Δ R[ κ ] =
  RowType Δ (λ Δ' → SemType Δ' κ) R[ κ ]
```

Types at ground kind \star and L are simply interpreted as `NormalTypes`. We interpret arrow-kinded types as *Kripke function spaces*, which permit the application of interpreted function ϕ at any environment Δ_2 provided a renaming from Δ_1 into Δ_2 . Note that we are defining `SemType` recursively (not inductively), and so the negative occurrence of `SemType` Δ_2 κ_1 is not a problem.

```
Renaming Δ1 Δ2 = TVar Δ1 κ → TVar Δ2 κ
KripkeFunction : Env → Kind → Kind → Set
KripkeFunction Δ1 κ1 κ2 = ∀ {Δ2} →
  Renaming Δ1 Δ2 → SemType Δ2 κ1 → SemType Δ2 κ2
```

The first three equations thus far are standard for this style of Agda mechanization, borrowing from Chapman et al. [2019]. Novel to our development is the interpretation of row-kinded types. First, we define the interpretation of row literals as finitely indexed maps to label-type pairs. (Here the type `Label` is a synonym for `String`, but could be any type with decidable equality and a strict total-order.)

```
Row : Set → Set
Row A = ∃[ n ] (Fin n → Label × A)
```

Next, we define a `RowType` inductively as one of four cases: either a row literal constructed by `row`, a neutral-labeled row singleton constructed by `_▷_`, an inert map constructed by `_$ _`, or an inert row complement constructed by `_⊔_`.

```
data RowType (Δ : Env)
  (T : Env → Set) : Kind → Set where
  row      : (ρ : Row (T Δ)) →
    OrderedRow ρ →
    RowType Δ T R[ κ ]
  _▷_      : NeutralType Δ L →
    T Δ →
    RowType Δ T R[ κ ]
  _$ _    : (∀ {Δ'} →
    Renaming Δ Δ' →
    NeutralType Δ' κ1 →
    T Δ') →
    NeutralType Δ R[ κ1 ] →
    RowType Δ T R[ κ2 ]
```

```

 $\_ \_$  : ( $\rho_2 \rho_1$  : RowType  $\Delta$  T R[  $\kappa$  ])  $\rightarrow$ 
      {nor : NotRow  $\rho_2$  or notRow  $\rho_1$ }  $\rightarrow$ 
      RowType  $\Delta$  T R[  $\kappa$  ]

```

Care must be taken to explain some nuances of each constructor. First, the `row` and `_ _` constructors are each constrained by predicates. The `OrderedRow ρ` predicate asserts that ρ has its string labels totally and ascendingly ordered—guaranteeing that labels in the row are unique and that rows are definitionally equal modulo ordering. The `NotRow ρ` predicate asserts simply that ρ was *not* constructed by `row`. In other words, it is not a simple row. This is important, as the complement of two row literals should reduce to a `Row`, so we must disallow the formation of complements in which at least one of the operands is a literal.

The next set of nuances come from dancing around Agda's positivity checker. It would have been preferable for us to have rather written the `row` and `_$` constructors as follows:

```

row      : ( $\rho$  : Row (SemType  $\Delta$   $\kappa$ ))  $\rightarrow$ 
          OrderedRow  $\rho$   $\rightarrow$ 
          RowType  $\Delta$  T R[  $\kappa$  ]
_$      : ( $\forall \{\Delta'\}$   $\rightarrow$ 
          Renaming  $\Delta$   $\Delta'$   $\rightarrow$ 
          SemType  $\Delta'$   $\kappa_1$   $\rightarrow$ 
          SemType  $\Delta'$   $\kappa_2$   $\rightarrow$ 
          NeutralType  $\Delta$  R[  $\kappa_1$  ]  $\rightarrow$ 
          RowType  $\Delta$  T R[  $\kappa_2$  ])  $\rightarrow$ 

```

Such a definition would have necessarily made the types `RowType` and `SemType` mutually inductive-recursive. But this would run afoul of Agda's positivity checker for the following reasons:

1. in the constructor `row`, the input `Row (SemType Δ κ)` makes a recursive call to `SemType Δ κ` , where it's not clear (to Agda) that this is a strictly smaller recursive call. To get around this, we parameterize the `RowType` type by $T : \text{Env} \rightarrow \text{Set}$ so that we may enforce this recursive call to be structurally smaller—hence the definition of `SemType` at kind `R[κ]` passes the argument $(\lambda \Delta' \rightarrow \text{SemType } \Delta' \kappa)$, which varies in environment but is at a strictly smaller kind.
2. The `_$` constructor takes a `KripkeFunction` as input, in which `SemType Δ' κ_1` occurs negatively, which Agda must outright reject. Here we borrow some clever machinery from Allais et al. [2013] and instead make the `KripkeFunction` accept the input `NeutralType Δ' κ_1` , which is already defined. The trick is that, as we will show in the next section, every `NeutralType` may be promoted to a `SemType`. In practice this is sufficient for our needs.

4.2 Reflection & reification

We have now declared three domains: the syntax of types, the syntax of normal and neutral types, and the embedded

domain of semantic types. Normalization by evaluation involves producing a *reflection* from neutral types to semantic types, a *reification* from semantic types to normal types, and an *evaluation* from types to semantic types. It follows thereafter that normalization is the reification of evaluation. Because we reason about types modulo η -expansion, reflection and reification are necessarily mutually recursive. (This is not the case however with e.g. Chapman et al. [2019].)

We describe the reflection logic before reification. Types at kind \star and L can be promoted straightforwardly with the `ne` constructor. Neutral types at arrow kind must be expanded into Kripke functions. Note that the input v has type `SemType Δ κ_1` and must be reified; additionally, τ is kinded in environment Δ_1 and so must be renamed to Δ_2 , the environment of v . The syntax `.` is used to construct an application of a `neutralType` to a `normalType`. Finally, a neutral row (e.g., a row variable) must be expanded into an inert mapping by $(\lambda r n \rightarrow \text{reflect } n)$, which is effectively the identity function.

```

reflect : NeutralType  $\Delta$   $\kappa$   $\rightarrow$  SemType  $\Delta$   $\kappa$ 
reify   : SemType  $\Delta$   $\kappa$   $\rightarrow$  NormalType  $\Delta$   $\kappa$ 

```

```

reflect { $\kappa = \star$ }  $\tau = \text{ne } \tau$ 
reflect { $\kappa = L$ }  $\tau = \text{ne } \tau$ 
reflect { $\kappa = \kappa_1 \rightarrow \kappa_2$ } =
   $\lambda r v \rightarrow \text{reflect } ((\text{rename } r \tau) \cdot \text{reify } v)$ 
reflect { $\kappa = R[ \kappa ]$ }  $\rho = (\lambda r n \rightarrow \text{reflect } n) \$ \rho$ 

```

The definition of reification is a little more involved. The first two equations are expected (τ is already in normal form). Functions are reified effectively by *eta*-expansion; note that we are using intrinsically-scoped De Bruijn variables, so Z constructs the zero'th variable and S induces a renaming in which each variable is incremented by one. (Recall that ϕ is a Kripke function space and so expects a renaming as argument.) The constructor ``` promotes a type variable to a `neutralType`. The equation of interest is in reifying rows. First note that we construct row literals at type `NormalType` likewise via the `row` constructor, which expects a proof that the row is well-ordered. Such a proof is given by the auxiliary lemma `reifyPreservesOrdering`, which proves what it says. Next, we use a helper function `reifyRow` to recursively build a list of `Label-NormalType` pairs (that is, the form of `NormalType` rows) from a semantic row. The empty case is trivial; the successor case must inspect the head of the list by inspecting `P fzero`, i.e., the label-type association of the zero'th finite index. From there we yield a semantic type τ which we reify and append to the rest of the list, built recursively.

```

reify { $\kappa = \star$ }  $\tau = \tau$ 
reify { $\kappa = L$ }  $\tau = \tau$ 
reify { $\kappa = \kappa_1 \rightarrow \kappa_2$ }  $\phi = ` \lambda (\text{reify } (\phi S (` Z)))$ 
reify { $\kappa = R[ \kappa ]$ } ( $\text{row } \rho q$ ) =
   $\text{row } (\text{reifyRow } \rho) (\text{reifyPreservesOrdering } q)$ 
where

```

```

reifyRow : Row (SemType  $\Delta$   $\kappa$ )  $\rightarrow$ 
  List (Label  $\times$  NormalType  $\Delta$   $\kappa$ )
reifyRow ( $\emptyset$  , P) = []
reifyRow (suc n , P) with P fzero
... | (1 ,  $\tau$ ) =
  (1 , reify  $\tau$ ) :: reifyRow (n , P  $\circ$  fsuc)

```

Finally, we have asserted that types are reduced modulo β -reduction and η -expansion. It follows that a given `NeutralType` should, after reflection and reification, end up in an expanded form. This is precisely how we define the promotion of `NeutralTypes` to `NormalTypes`:

```

 $\eta$ -norm : NeutralType  $\Delta$   $\kappa$   $\rightarrow$  NormalType  $\Delta$   $\kappa$ 
 $\eta$ -norm = reify  $\circ$  reflect

```

This function is necessary: the `NormalType` constructor `ne` stipulates that we may only promote neutral derivations to normal derivations at *ground kind* (rule (κ_{nf-NE})). Hence η -norm is the only means by which we may promote neutral types at row or arrow kind.

4.3 Helping evaluation

We will build our evaluation function incrementally; we find it clearer to incrementally build helpers for sub-computation (e.g., mapping or the complement) on our way up to full evaluation. We describe these helpers next.

4.3.1 Semantic application. We first define semantic application straightforwardly as Agda application under the identity renaming.

```

_·'_ : SemType  $\Delta$  ( $\kappa_1 \rightarrow \kappa_2$ )  $\rightarrow$ 
  SemType  $\Delta$   $\kappa_1$   $\rightarrow$ 
  SemType  $\Delta$   $\kappa_2$ 
 $\phi$  ·' v =  $\phi$  id v

```

4.3.2 Semantic mapping. Mapping over rows is a form of computation novel to $R\omega\mu$'s equivalence relation. We define the mapping ϕ \$ ρ over the four cases a semantic row may take. When ρ is neutral-labeled, we simply apply ϕ to its contents. The case where ρ is a row literal is interesting in that our choice of representation for row literals as Agda functions comes to pay off: we may express the mapping of ϕ across the row (n , P) by pre-composing P with ϕ (note that we must appropriately `fmap` ϕ over the pair's second component).

```

_$_'_ : SemType  $\Delta$  ( $\kappa_1 \rightarrow \kappa_2$ )  $\rightarrow$ 
  SemType  $\Delta$  R[  $\kappa_1$  ]  $\rightarrow$ 
  SemType  $\Delta$  R[  $\kappa_2$  ]
 $\phi$  $' (1  $\triangleright$   $\tau$ ) = 1  $\triangleright$  ( $\phi$  ·'  $\tau$ )
 $\phi$  $' (row (n , P) q) = row (n , fmap ( $\phi$  id)  $\circ$  P)

```

4.4 Evaluation

Evaluation warrants an environment that maps type variables to semantic types. The identity environment, which fixes the meaning of variables, is given as the composition of reflection and \sim , the constructor of `NeutralTypes` from `TVars`.

```

SemEnv : Env  $\rightarrow$  Env  $\rightarrow$  Set
SemEnv  $\Delta_1$   $\Delta_2$  = TVar  $\Delta_1$   $\rightarrow$  SemType  $\Delta_2$   $\kappa$ 
idEnv : SemEnv  $\Delta$   $\Delta$ 
idEnv = reflect  $\circ$  `

```

4.5 Normalization

Normalization in the NbE approach is simply the composition of reification after evaluation.

```

 $\Downarrow$  : Type  $\Delta$   $\kappa$   $\rightarrow$  NormalType  $\Delta$   $\kappa$ 
 $\Downarrow$   $\tau$  = reify (eval  $\tau$  idEnv)

```

It will be helpful in the coming metatheory to define an inverse embedding by induction over the `NormalType` structure. The definition is entirely expected and omitted.

```

 $\Uparrow$  : NormalType  $\Delta$   $\kappa$   $\rightarrow$  Type  $\Delta$   $\kappa$ 

```

5 Mechanized metatheory

This section gives a deeper exposition on the metatheory summarized (§3.2). We forego syntactic tying of claims and give a deeper explanation of the proof techniques involved.

5.1 Stability

Stability follows by simple induction on typing derivations.

Theorem 5.1 (stability).

Stability implies surjectivity and idempotency. Dual to surjectivity, stability also implies that embedding is injective.

5.2 A logical relation for completeness

5.2.1 Properties.

5.2.2 Logical environments.

5.2.3 The fundamental theorem and completeness.

5.3 A logical relation for soundness

5.3.1 Properties.

5.3.2 Logical environments.

5.3.3 The fundamental theorem and Soundness.

6 Most closely related work

References

- Guillaume Allais, Pierre Boutillier, and Conor McBride. New equations for neutral terms: A sound and complete decision procedure, formalized, 2013. URL <https://arxiv.org/abs/1304.0809>.
- James Chapman, Roman Kireev, Chad Nester, and Philip Wadler. System F in agda, for fun and profit. In Graham Hutton, editor, *Mathematics of Program Construction - 13th International Conference, MPC 2019, Porto, Portugal, October 7-9, 2019, Proceedings*, volume 11825 of *Lecture Notes in Computer Science*, pages 255–297. Springer, 2019. ISBN 978-3-030-33635-6. doi: 10.1007/978-3-030-33636-3_10. URL https://doi.org/10.1007/978-3-030-33636-3_10.
- Alex Hubers and J. Garrett Morris. Generic programming with extensible data types: Or, making ad hoc extensible data types less ad hoc. *Proc. ACM Program. Lang.*, 7(ICFP):356–384, 2023. doi: 10.1145/3607843. URL <https://doi.org/10.1145/3607843>.

771	Alex Hubers, Apoorv Ingle, Andrew Marmaduke, and J. Garrett Morris.	abs/2410.11742.	826
772	Extensible recursive functions, algebraically, 2024. URL https://arxiv.org/		827
773			828
774			829
775			830
776			831
777			832
778			833
779			834
780			835
781			836
782			837
783			838
784			839
785			840
786			841
787			842
788			843
789			844
790			845
791			846
792			847
793			848
794			849
795			850
796			851
797			852
798			853
799			854
800			855
801			856
802			857
803			858
804			859
805			860
806			861
807			862
808			863
809			864
810			865
811			866
812			867
813			868
814			869
815			870
816			871
817			872
818			873
819			874
820			875
821			876
822			877
823			878
824			879
825			880