

Type Normalization in $R\omega\mu$

ALEX HUBERS, The University of Iowa, USA

1 INTRODUCTION

We describe the normalization-by-evaluation (NBE) of types in $R\omega\mu$. Types are normalized modulo β - and η -equivalence—that is, to $\beta\eta$ -long forms. Because the type system of $R\omega\mu$ is a strict extension of System $F\omega\mu$, type level computation for arrow kinds is isomorphic to reduction of arrow types in the STLC. Novel to this report are the reductions of Π , Σ , and label bound terms.

2 SYNTAX OF KINDS

Our formalization of $R\omega\mu$ types is *intrinsic*, meaning we define the syntax of *typing* and *kinding judgments*, foregoing any description of untyped syntax. The syntax of types is indexed by kinding environments and kinds, defined below.

```
data Kind : Set where
  ★      : Kind
  L      : Kind
  _'→_   : Kind → Kind → Kind
  R[_]   : Kind → Kind

infixr 5 _'→_
```

The kind system of $R\omega\mu$ defines \star as the type of types; L as the type of labels; (\rightarrow) as the type of type operators; and $R[\kappa]$ as the type of *rows* containing types at kind κ . As shorthand, we write $R^n[\kappa]$ to denote n repeated applications of R to the type κ —e.g., $R^3[\kappa]$ is shorthand for $R[R[R[\kappa]]]$.

The syntax of kinding environments is given below. Kinding environments are isomorphic to lists of kinds.

```
data KEnv : Set where
  ε : KEnv
  _»_ : KEnv → Kind → KEnv
```

Let the metavariables Δ and κ range over kinding environments and kinds, respectively. Correspondingly, we define *generalized variables* in Agda at these names.

```
private
variable
  Δ Δ1 Δ2 Δ3 : KEnv
  κ κ1 κ2 : Kind
```

The syntax of intrinsically well-scoped De-Bruijn type variables is given below. We say that the kind variable x is indexed by kinding environment Δ and kind κ to specify that x has kind κ in kinding environment Δ .

Author's address: [Alex Hubers](#), Department of Computer Science, The University of Iowa, 14 MacLean Hall, Iowa City, Iowa, USA, alexander-hubers@uiowa.edu.

```

50 data KVar : KEnv → Kind → Set where
51   Z : KVar (Δ „ κ) κ
52   S : KVar Δ κ1 → KVar (Δ „ κ2) κ1
53
54
55

```

3 SYNTAX OF TYPES

$R\omega\mu$ is a qualified type system with predicates of the form $\rho_1 \lesssim \rho_2$ and $\rho_1 \cdot \rho_2 \sim \rho_3$ for row-kinded types ρ_1 , ρ_2 , and ρ_3 . Because predicates occur in types and types occur in predicates, the syntax of well-kinded types and well-kinded predicates are mutually recursive. The syntax for each is given below. we describe (in this order) the syntactic components belonging to System $F\omega\mu$, qualified type systems, and system $R\omega$.

```

61 data Pred (Δ : KEnv) : Kind → Set
62 data Type Δ : Kind → Set
63 data Type Δ where
64
65
66

```

```

65   ‘ :
66     (α : KVar Δ κ) →
67     Type Δ κ
68
69

```

```

69   ‘λ :
70     (τ : Type (Δ „ κ1) κ2) →
71     Type Δ (κ1 ‘→ κ2)
72
73

```

```

73   _·_ :
74     (τ1 : Type Δ (κ1 ‘→ κ2)) →
75     (τ2 : Type Δ κ1) →
76     Type Δ κ2
77
78

```

```

78   _‘→_ :
79     (τ1 : Type Δ ★) →
80     (τ2 : Type Δ ★) →
81     Type Δ ★
82
83

```

```

83   ‘∀ :
84     (τ : Type (Δ „ κ) ★) →
85     Type Δ ★
86
87

```

```

87   μ :
88     (F : Type Δ (★ ‘→ ★)) →
89     Type Δ ★
90
91

```

The first three constructors are analogous to the terms of the STLC. the constructor \sim classifies term functions; the constructor \forall classifies type-in-term quantification; and the constructor μ classifies recursive terms. Note that μ could be further generalized to kind $\kappa \sim \rightarrow \star$; however, we found that kind $\star \sim \rightarrow \star$ was sufficient for our needs while simplifying both presentation and mechanization.

The syntax of qualified types is given below.

```

99   $\Rightarrow$  :
100  ( $\pi$  :  $\text{Pred } \Delta \text{ R}[ \kappa_1 ]$ )  $\rightarrow$  ( $\tau$  :  $\text{Type } \Delta \star$ )  $\rightarrow$ 
101   $\text{Type } \Delta \star$ 
102

```

The type $\pi \Rightarrow \tau$ states that τ is *qualified* by the predicate π —that is, the type variables bound in τ are restricted in instantiation to just those that satisfy the predicate π . This is completely analogous to identical syntax used in Haskell to introduce typeclass qualification. Predicates are defined below (after the presentation of type syntax).

We now describe the syntax exclusive to $R\omega\mu$, beginning with label kind introduction and elimination. Labels are first-class entities in $R\omega\mu$, and may be represented by both constants and variables.

```

111 lab :
112  ( $l$  :  $\text{Label}$ )  $\rightarrow$ 
113   $\text{Type } \Delta \text{ L}$ 
114
115  $\llbracket \_ \rrbracket$  :
116  ( $\tau$  :  $\text{Type } \Delta \text{ L}$ )  $\rightarrow$ 
117   $\text{Type } \Delta \star$ 
118

```

Label constants in $R\omega\mu$ are constructed from the type Label ; in our mechanization, Label is a type synonym for String , but one could choose any other candidate with decidable equality. Types at label kind L may be cast to *label singletons* by the $\llbracket _ \rrbracket$ constructor. This makes labels first-class entities: for example, as the type $\llbracket \text{lab } "1" \rrbracket$ has kind \star , it can be inhabited by a term.

Types at row kind are constructed by one of the following three constructors.

```

125  $\epsilon$  :
126   $\text{Type } \Delta \text{ R}[ \kappa ]$ 
127
128  $\triangleright$  :
129  ( $l$  :  $\text{Type } \Delta \text{ L}$ )  $\rightarrow$  ( $\tau$  :  $\text{Type } \Delta \kappa$ )  $\rightarrow$ 
130   $\text{Type } \Delta \text{ R}[ \kappa ]$ 
131
132  $\<\$>$  :
133  ( $f$  :  $\text{Type } \Delta (\kappa_1 \rightarrow \kappa_2)$ )  $\rightarrow$  ( $\tau$  :  $\text{Type } \Delta \text{ R}[ \kappa_1 ]$ )  $\rightarrow$ 
134   $\text{Type } \Delta \text{ R}[ \kappa_2 ]$ 
135

```

Rows in $R\omega\mu$ are either the empty row ϵ , a labeled row ($1 \triangleright \tau$), or a row mapping $f \<\$> \tau$. The row mapping $f \<\$> (1 \triangleright \tau)$ describes the lifting of the function f over row ($1 \triangleright \tau$), which we will define to equal $(1 \triangleright f \tau)$ in the case where the right hand applicand is a labeled row. We will show that rows in Rome (that is, types at row kind) reduce to either the empty row ϵ or a labeled row ($1 \triangleright \tau$) after normalization. There are two important consequences of this canonicity: firstly, we treat row mapping $_ \<\$> _$ as having latent computation to perform (there are no normal types with form $f \<\$> \tau$ except when τ is a neutral variable). The second consequence is that we do not permit the formation of rows with more than one label-type association. Such rows are instead formed as type variables with predicates specifying the shape of the row.

Rows in $R\omega\mu$ are eliminated by the Π and Σ constructors.

```

148  $\Pi :$ 
149    $\text{Type } \Delta (R[\kappa] \multimap \kappa)$ 
150
151  $\Sigma :$ 
152    $\text{Type } \Delta (R[\kappa] \multimap \kappa)$ 

```

Given a type ρ at row kind, $\Pi\rho$ constructs a record with label-type associations from ρ and $\Sigma\rho$ constructs a variant that has label and type from ρ . We choose to represent Π and Σ as type constants at kind $(R[\kappa] \multimap \kappa)$; we will show that many applications of Π and Σ induce type reductions, and hence it is convenient to group such reductions with type application.

The syntax of predicates is given below. The predicate $\rho_1 \lesssim \rho_2$ states that label-to-type mappings in ρ_1 are a subset of those in ρ_2 ; the predicate $\rho_1 \cdot \rho_2 \sim \rho_3$ states that the combination of mappings in ρ_1 and ρ_2 equals ρ_3 .

```

162 data Pred  $\Delta$  where
163    $\_ \cdot \_ \sim \_ :$ 
164      $(\rho_1 \rho_2 \rho_3 : \text{Type } \Delta R[\kappa]) \rightarrow$ 
165      $\text{Pred } \Delta R[\kappa]$ 
166
167    $\_ \lesssim \_ :$ 
168      $(\rho_1 \rho_2 : \text{Type } \Delta R[\kappa]) \rightarrow$ 
169      $\text{Pred } \Delta R[\kappa]$ 

```

Hubers and Morris [2023] implicitly define two type-level row lifting operators, *left mapping* $\langle \$ \rangle$ and *right mapping* $\langle ? \rangle$, but the latter is superfluous. We appeal to the kinds of these operators for their intuition: left mapping $f \langle \$ \rangle \rho$ lifts a function at arrow kind $f : \kappa_1 \rightarrow \kappa_2$ into a function at kind $R[\kappa_1] \rightarrow R[\kappa_2]$ and then applies it to $\rho : R[\kappa_2]$. We may define right mapping (named *flap* and written $\langle ? \rangle$, after similar Haskell operators) of row function $f : R[\kappa_1 \rightarrow \kappa_2]$ over type $\tau : \kappa_1$ using left mapping under the following identity:

$$f \langle ? \rangle \tau = (\lambda g. g \tau) \langle \$ \rangle f$$

which we encode in Agda as follows:

```

182 flap : Type  $\Delta (R[\kappa_1 \multimap \kappa_2] \multimap \kappa_1 \multimap R[\kappa_2])$ 
183 flap = 'λ ('λ (('λ (('Z) · ('(S Z)))) <$> ('(S Z))))
184
185  $\_ \langle ? \rangle \_ :$  Type  $\Delta (R[\kappa_1 \multimap \kappa_2]) \rightarrow \text{Type } \Delta \kappa_1 \rightarrow \text{Type } \Delta R[\kappa_2]$ 
186  $f \langle ? \rangle a = \text{flap } f \cdot a$ 

```

(We choose to define $_ \langle ? \rangle _$ as the application of *flap* to inputs f and a so that we needn't pollute the definition with weakenings of its arguments.)

3.1 Type renaming

We closely follow Wadler et al. [2022] and Chapman et al. [2019] in defining a *type renaming* as a function from type variables in one kinding environment to type variables in another. This is the *parallel renaming and substitution* approach for which weakening and single variable substitution are special cases. The code we establish now will be mimicked again for both normal types and for

terms; many names are reused, and so we find it helpful to index duplicate names by a suffix. The suffix $_k$ specifies that this definition describes the Type syntax.

$\text{Renaming}_k : \text{KEnv} \rightarrow \text{KEnv} \rightarrow \text{Set}$
 $\text{Renaming}_k \Delta_1 \Delta_2 = \forall \{\kappa\} \rightarrow \text{KVar} \Delta_1 \kappa \rightarrow \text{KVar} \Delta_2 \kappa$

We will let the metavariable ρ range over both renamings and types at row kind.

Lifting can be thought of as the weakening of a renaming, and permits renamings to be pushed under binders.

$\text{lift}_k : \text{Renaming}_k \Delta_1 \Delta_2 \rightarrow \text{Renaming}_k (\Delta_1 \text{ ,, } \kappa) (\Delta_2 \text{ ,, } \kappa)$
 $\text{lift}_k \rho \text{ Z} = \text{Z}$
 $\text{lift}_k \rho (S \ x) = S (\rho \ x)$

We define renaming as a function that translates a kinding derivation in kinding environment Δ_1 to environment Δ_2 provided a renaming from Δ_1 to Δ_2 . The definition proceeds by induction on the input kinding derivation. In the variable case, we use ρ to rename variable x . In the λ and \forall cases, we must lift the renaming ρ over the type variable introduced by these binders. The rest of the cases are effectively just congruence over the type structure.

$\text{ren}_k : \text{Renaming}_k \Delta_1 \Delta_2 \rightarrow \text{Type} \Delta_1 \kappa \rightarrow \text{Type} \Delta_2 \kappa$
 $\text{renPred}_k : \text{Renaming}_k \Delta_1 \Delta_2 \rightarrow \text{Pred} \Delta_1 R[\kappa] \rightarrow \text{Pred} \Delta_2 R[\kappa]$
 $\text{ren}_k \rho (\text{' } x) = \text{' } (\rho \ x)$
 $\text{ren}_k \rho (\text{' } \lambda \ \tau) = \text{' } \lambda (\text{ren}_k (\text{lift}_k \rho) \ \tau)$
 $\text{ren}_k \rho (\pi \Rightarrow \tau) = \text{renPred}_k \rho \ \pi \Rightarrow \text{ren}_k \rho \ \tau$
 $\text{ren}_k \rho (\text{' } \forall \ \tau) = \text{' } \forall (\text{ren}_k (\text{lift}_k \rho) \ \tau)$
 $\text{ren}_k \rho \ \epsilon = \epsilon$
 $\text{ren}_k \rho (\tau_1 \cdot \tau_2) = (\text{ren}_k \rho \ \tau_1) \cdot (\text{ren}_k \rho \ \tau_2)$
 $\text{ren}_k \rho (\tau_1 \text{' } \rightarrow \tau_2) = (\text{ren}_k \rho \ \tau_1) \text{' } \rightarrow (\text{ren}_k \rho \ \tau_2)$
 $\text{ren}_k \rho (\mu \ F) = \mu (\text{ren}_k \rho \ F)$
 $\text{ren}_k \rho \ \Pi = \Pi$
 $\text{ren}_k \rho \ \Sigma = \Sigma$
 $\text{ren}_k \rho (\text{lab } x) = \text{lab } x$
 $\text{ren}_k \rho (l \triangleright \tau) = \text{ren}_k \rho \ l \triangleright \text{ren}_k \rho \ \tau$
 $\text{ren}_k \rho [\ell] = [\text{ren}_k \rho \ \ell]$
 $\text{ren}_k \rho (f \text{ <$> } m) = \text{ren}_k \rho \ f \text{ <$> } \text{ren}_k \rho \ m$

As Type and Pred are mutually inductive, we must define renPred_k as mutually recursive to ren_k . Its definition is completely unsurprising.

$\text{renPred}_k \rho (\rho_1 \cdot \rho_2 \sim \rho_3) = \text{ren}_k \rho \ \rho_1 \cdot \text{ren}_k \rho \ \rho_2 \sim \text{ren}_k \rho \ \rho_3$
 $\text{renPred}_k \rho (\rho_1 \lesssim \rho_2) = (\text{ren}_k \rho \ \rho_1) \lesssim (\text{ren}_k \rho \ \rho_2)$

Finally, weakening is a special case of renaming.

$\text{weaken}_k : \text{Type} \Delta \kappa_2 \rightarrow \text{Type} (\Delta \text{ ,, } \kappa_1) \kappa_2$
 $\text{weaken}_k = \text{ren}_k \ S$

3.2 Type substitution

We wish to give both a declarative and algorithmic treatment of type equivalence. For the latter, we will normalize types to normal forms, meaning types are equivalent iff their normal forms are definitionally equal. For the former, we must define β -substitution syntactically so that we can express β -equivalence of types declaratively. In our development, β -reduction is a special case of substitution.

We define a substitution as a function mapping type variables in context Δ_1 to types in context Δ_2 .

$\text{Substitution}_k : \text{KEnv} \rightarrow \text{KEnv} \rightarrow \text{Set}$

$\text{Substitution}_k \Delta_1 \Delta_2 = \forall \{\kappa\} \rightarrow \text{KVar } \Delta_1 \kappa \rightarrow \text{Type } \Delta_2 \kappa$

Substitutions must be lifted over binders, just as is done for renamings.

$\text{lifts}_k : \text{Substitution}_k \Delta_1 \Delta_2 \rightarrow \text{Substitution}_k (\Delta_1 \gg \kappa) (\Delta_2 \gg \kappa)$

$\text{lifts}_k \sigma \text{Z} = \text{'Z}$

$\text{lifts}_k \sigma (\text{S } x) = \text{weaken}_k (\sigma x)$

Substitution is defined inductively over types in a similar fashion to renaming. Note that this is *simultaneous* substitution and renaming—The variable case translates type variable x to the type σx , for which the substitution σ also performs a renaming from environment Δ_1 to Δ_2 . The rest of the cases (as with renaming) are either congruences over the type structure or congruences plus lifting of the substitution. Again, substitution over predicates is defined mutually recursively.

$\text{sub}_k : \text{Substitution}_k \Delta_1 \Delta_2 \rightarrow \text{Type } \Delta_1 \kappa \rightarrow \text{Type } \Delta_2 \kappa$

$\text{subPred}_k : \text{Substitution}_k \Delta_1 \Delta_2 \rightarrow \text{Pred } \Delta_1 \kappa \rightarrow \text{Pred } \Delta_2 \kappa$

$\text{sub}_k \sigma \epsilon = \epsilon$

$\text{sub}_k \sigma (\text{' } x) = \sigma x$

$\text{sub}_k \sigma (\text{' } \lambda \tau) = \text{' } \lambda (\text{sub}_k (\text{lifts}_k \sigma) \tau)$

$\text{sub}_k \sigma (\tau_1 \cdot \tau_2) = (\text{sub}_k \sigma \tau_1) \cdot (\text{sub}_k \sigma \tau_2)$

$\text{sub}_k \sigma (\tau_1 \text{' } \rightarrow \tau_2) = (\text{sub}_k \sigma \tau_1) \text{' } \rightarrow (\text{sub}_k \sigma \tau_2)$

$\text{sub}_k \sigma (\pi \Rightarrow \tau) = \text{subPred}_k \sigma \pi \Rightarrow \text{sub}_k \sigma \tau$

$\text{sub}_k \sigma (\text{' } \forall \tau) = \text{' } \forall (\text{sub}_k (\text{lifts}_k \sigma) \tau)$

$\text{sub}_k \sigma (\mu F) = \mu (\text{sub}_k \sigma F)$

$\text{sub}_k \sigma (\Pi) = \Pi$

$\text{sub}_k \sigma \Sigma = \Sigma$

$\text{sub}_k \sigma (\text{lab } x) = \text{lab } x$

$\text{sub}_k \sigma (l \triangleright \tau) = \text{sub}_k \sigma l \triangleright \text{sub}_k \sigma \tau$

$\text{sub}_k \sigma [\ell] = [\text{sub}_k \sigma \ell]$

$\text{sub}_k \sigma (f \text{' } \<\$> a) = \text{sub}_k \sigma f \text{' } \<\$> \text{sub}_k \sigma a$

$\text{subPred}_k \sigma (\rho_1 \cdot \rho_2 \sim \rho_3) = \text{sub}_k \sigma \rho_1 \cdot \text{sub}_k \sigma \rho_2 \sim \text{sub}_k \sigma \rho_3$

$\text{subPred}_k \sigma (\rho_1 \lesssim \rho_2) = (\text{sub}_k \sigma \rho_1) \lesssim (\text{sub}_k \sigma \rho_2)$

We define the extension of a substitution σ by the type τ functionally. If we had chosen to represent a Substitution_k as a list, extension would be done by the cons constructor. In a De-Bruijn representation, the most recently appended variable is zero—hence an extension here maps the zero variable to τ in the Z case and maps each variable (S x) to its value in σ at predecessor x .

```

295 extendk : Substitutionk Δ1 Δ2 → (τ : Type Δ2 κ) → Substitutionk (Δ1 „ κ) Δ2
296 extendk σ τ Z = τ
297 extendk σ τ (S x) = σ x
298

```

Finally, β -substitution is simply a special case of substitution. Note that the constructor \cdot has type $KVar \Delta \kappa \rightarrow Type \Delta \kappa$, making it a substitution. It is in fact an identity substitution, which fixes the meaning of its type variables, hence it is the substitution we choose to extend when defining β -substitution.

```

303 _βk[_] : Type (Δ „ κ1) κ2 → Type Δ κ1 → Type Δ κ2
304 τ1 βk[ τ2 ] = subk (extendk ‘ τ2) τ1
305

```

4 TYPE EQUIVALENCE

We define type and predicate equivalence mutually recursively. You may think of type equivalence also as a sort of small-step relation on types, as we include rules to equate β -equivalent and η -equivalent types, as well as a number of computational steps a row kinded type may take.

```

311 infix 0 _≡t_
312 infix 0 _≡p_
313 data _≡p_ : Pred Δ R[ κ ] → Pred Δ R[ κ ] → Set
314 data _≡t_ : Type Δ κ → Type Δ κ → Set
315

```

Unless otherwise quantified, let the metavariable l range over types with label kind, let π range over predicates, and let τ and v range over types:

```

319 private
320   variable
321     l l1 l2 l3 : Type Δ L
322     ρ1 ρ2 ρ3 : Type Δ R[ κ ]
323     π1 π2 : Pred Δ R[ κ ]
324     τ τ1 τ2 τ3 v v1 v2 v3 : Type Δ κ
325

```

The rules for predicate equivalence are uninteresting: two predicates are considered equivalent when their component types are equivalent.

```

328 data _≡p_ where
329   _eq-≤_ :
330     τ1 ≡t v1 → τ2 ≡t v2 →
331     τ1 ≤ τ2 ≡p v1 ≤ v2
332
333   _eq-·~_ :
334     τ1 ≡t v1 → τ2 ≡t v2 → τ3 ≡t v3 →
335     τ1 · τ2 ~ τ3 ≡p v1 · v2 ~ v3
336

```

The first three rules for type equivalence state that it is an equivalence relation.

```

337
338 data _≡t_ where
339   eq-refl :
340     τ ≡t τ
341

```

eq-sym :

$$\tau_1 \equiv \tau_2 \rightarrow$$

$$\tau_2 \equiv \tau_1$$

eq-trans :

$$\tau_1 \equiv \tau_2 \rightarrow \tau_2 \equiv \tau_3 \rightarrow$$

$$\tau_1 \equiv \tau_3$$

Type equivalence is congruent over the total structure of types, including λ -bindings (hence you may view type normalization as being *call-by-value*). We omit the other eight congruence rules.

eq- λ : $\forall \{\tau \ v : \text{Type } (\Delta \text{ „ } \kappa_1) \ \kappa_2\} \rightarrow$

$$\tau \equiv v \rightarrow$$

$$' \lambda \ \tau \equiv ' \lambda \ v$$

We have one η -equivalence rule. It is henceforth useful to view the following rules as directed left-to-right, as normal forms are produced on the right-hand side.

eq- η : $\forall \{f : \text{Type } \Delta \ (\kappa_1 \rightarrow \kappa_2)\} \rightarrow$

$$f \equiv ' \lambda \ (\text{weaken}_k f \cdot (' \ Z))$$

The rules that remain as *computational*—these are precisely the rules we would use to define small-step reduction of types. We begin with the β -equivalence rule, which states that lambda abstractions applied to arguments are equivalent to their beta reduction.

eq- β : $\forall \{\tau_1 : \text{Type } (\Delta \text{ „ } \kappa_1) \ \kappa_2\} \{\tau_2 : \text{Type } \Delta \ \kappa_1\} \rightarrow$

$$((' \lambda \ \tau_1) \cdot \tau_2) \equiv (\tau_1 \ \beta_k [\ \tau_2 \])$$

The next two rules specify the computational behavior of mapping over rows. Rule (eq- ϵ) states that mapping over the empty row ϵ should yield the empty row; rule eq- \rightarrow states that mapping over a labeled row should push the left applicand into the body of the row.

eq- ϵ : $\{F : \text{Type } \Delta \ (\kappa_1 \rightarrow \kappa_2)\} \rightarrow$

$$(F \ \epsilon) \equiv \epsilon$$

eq- \rightarrow : $\forall \{l\} \{\tau : \text{Type } \Delta \ \kappa_1\} \{F : \text{Type } \Delta \ (\kappa_1 \rightarrow \kappa_2)\} \rightarrow$

$$(F \ \rightarrow (l \triangleright \tau)) \equiv (l \triangleright (F \cdot \tau))$$

We wish to establish that normal forms of types at row kind are either the empty row ϵ or labeled rows. This is, of course, not the case for types in general. For example, the type $\Pi \cdot (l \triangleright \tau)$ has row kind when τ has row kind $R[\ \kappa \]$. In this case, rule eq- Π pushes the Π over the label so that a canonical form is restored.

eq- Π : $\forall \{l\} \{\tau : \text{Type } \Delta \ R[\ \kappa \]\} \rightarrow$

$$\Pi \cdot (l \triangleright \tau) \equiv (l \triangleright (\Pi \cdot \tau))$$

The application of Π and Σ to a type τ at nested-row kind is in fact just the mapping of Π and Σ over τ :

eq- Π : $\forall \{\tau : \text{Type } \Delta \ R[\ R[\ \kappa \]]\} \rightarrow$

$$\Pi \cdot \tau \equiv \Pi \ \epsilon \ \$ \ \tau$$

Likewise to rows, we wish to show that normal forms of types at arrow kind are canonically λ -bound. However, the type $\Pi \cdot (1 \triangleright \lambda \tau)$ has arrow kind! Rule $\text{eq-}\Pi\lambda$ pushes the λ outwards in order to restore canonicity and so that application of $\Pi \cdot (1 \triangleright \lambda \tau)$ to an applicand is simply β -reduction.

```

eq- $\Pi\lambda$  :  $\forall \{l\} \{\tau : \text{Type } (\Delta \text{ ,, } \kappa_1) \kappa_2\} \rightarrow$ 
   $\Pi \cdot (l \triangleright \lambda \tau) \equiv \lambda (\Pi \cdot (\text{weaken}_\kappa l \triangleright \tau))$ 

```

Finally, in many cases (such as record concatenation and variant branching) it is necessary to reassociate the application $(\Pi \rho) \tau$ inward so that Π (or Σ) are the outermost syntax. We observe the following reassociation identity:

```

eq- $\Pi$ -assoc :  $\forall \{\rho : \text{Type } \Delta (R[\kappa_1 \xrightarrow{\quad} \kappa_2])\} \{\tau : \text{Type } \Delta \kappa_1\} \rightarrow$ 
   $(\Pi \cdot \rho) \cdot \tau \equiv \Pi \cdot (\rho \text{ <?> } \tau)$ 

```

The definition of $_ \equiv _$ concludes by repeating the last four rules, replacing each Π with Σ . As a final aside, it might be thought that we could have rid ourselves of the syntax for mapping by elaborating types at kind $R[\kappa_1 \rightarrow \kappa_2]$. For example, the type $(1 \triangleright \lambda x : \kappa_1. \tau)$ could perhaps have its λ binding pushed outside to yield $\lambda x : \kappa_1. (1 \triangleright \tau)$. However, this would not be kind-preserving (the latter has kind $\kappa_1 \rightarrow R[\kappa_2]$), and therefore such a translation would induce a normalization that does not preserve kinds. We believe it would be possible but complicated to consider a kind-changing translation.

5 NORMAL TYPES

As is common in other *normalization by evaluation* approaches, we separate *neutral types* from *normal types*. These two definitions are defined mutually inductively with the data type for normal predicates:

```

data NormalType ( $\Delta : \text{KEnv}$ ) : Kind  $\rightarrow$  Set
data NormalPred ( $\Delta : \text{KEnv}$ ) : Kind  $\rightarrow$  Set
data NeutralType  $\Delta$  : Kind  $\rightarrow$  Set

```

A type is neutral if it is (respectively) (i) a variable, (ii) the application of a variable to an argument, or (iii) the mapping of a normal function type over a neutral row type. Intuitively, neutral forms are forms for which computation is "stuck" waiting on a variable to be substituted for a canonical form. Note that this third neutral form (row mapping) is novel to our development, and, in comparison to application, inverts the normal/neutral expectation of its arguments. It captures the stuck nature of a type such as $(1 \triangleright \lambda x. M) \text{ <\$> } \rho$ —that is, we are unable to map a function over a type variable.

```

data NeutralType  $\Delta$  where

```

```

  ' :

```

```

    ( $\alpha : \text{KVar } \Delta \kappa$ )  $\rightarrow$ 
    NeutralType  $\Delta \kappa$ 

```

```

  '·' :

```

```

    ( $f : \text{NeutralType } \Delta (\kappa_1 \xrightarrow{\quad} \kappa)$ )  $\rightarrow$ 
    ( $\tau : \text{NormalType } \Delta \kappa_1$ )  $\rightarrow$ 
    NeutralType  $\Delta \kappa$ 

```

```

  '·<\$>' :

```

```

(F : NormalType Δ (κ1 '→ κ2)) → (τ : NeutralType Δ R[ κ1 ]) →
NeutralType Δ (R[ κ2 ])

```

A predicate is normal if its component types are each normal.

```

data NormalPred Δ where
  _·~_ :
    (ρ1 ρ2 ρ3 : NormalType Δ R[ κ ]) →
    NormalPred Δ R[ κ ]
  _≤_ :
    (ρ1 ρ2 : NormalType Δ R[ κ ]) →
    NormalPred Δ R[ κ ]

```

Because we consider the normalization of types modulo η -equivalence, we wish to restrict our normal types to η -long form. This can be done by restricting the construction of normal-neutral types to just ground kind. This also ensures a canonical form for arrow-kinded normal types, as neutral types at arrow-kind cannot be promoted to normal types. We define a Ground predicate on types that maps all non-arrow kinds to the unit type \top and maps the arrow kind to \perp . (In other words, Ground κ is trivially inhabitable so long as $\kappa \neq \kappa_1 \rightarrow \kappa_2$.)

```

Ground : Kind → Set
Ground ★ = ⊤
Ground L = ⊤
Ground (κ '→ κ1) = ⊥
Ground R[ κ ] = ⊤

```

It is easy to show that this predicate is decidable.

```

ground? : ∀ κ → Dec (Ground κ)
ground? ★ = yes tt
ground? L = yes tt
ground? (κ '→ κ1) = no (λ ())
ground? R[ κ ] = yes tt

```

Now we may restrict the ne constructor to promoting just neutral types at ground kind by adding the (implicit) requirement that ne only be used when Ground κ is satisfied. To make this evidence easy to populate when κ is known, we employ a well-known proof-by-reflection trick (see [Wadler et al. \[2022\]](#)) and require evidence of the form True (ground? κ).

```

data NormalType Δ where
  ne :
    (x : NeutralType Δ κ) → {ground : True (ground? κ)} →
    NormalType Δ κ

```

Likewise, to ensure canonical forms of rows, we restrict Π and Σ to formation at kind \star and L. The constructors for record types are given below.

```

491  $\Pi :$ 
492    $(\rho : \text{NormalType } \Delta \text{ R[ } \star \text{ ]}) \rightarrow$ 
493    $\text{NormalType } \Delta \star$ 

```

```

494  $\Pi L :$ 
495    $(\rho : \text{NormalType } \Delta \text{ R[ L ]}) \rightarrow$ 
496    $\text{NormalType } \Delta L$ 

```

The rest of the `NormalType` syntax is identical to the `Type` syntax with the exception that we remove the ``` constructor for variables and Π and Σ constructors at arbitrary kind. We choose not to omit this syntax, as our proofs of canonicity follow from knowing the totality of `NormalType` constructors.

```

503 -  $F\omega$ 

```

```

504  $\lambda :$ 
505    $(\tau : \text{NormalType } (\Delta \text{ „ } \kappa_1) \kappa_2) \rightarrow$ 
506    $\text{NormalType } \Delta (\kappa_1 \text{ ' } \rightarrow \kappa_2)$ 

```

```

507  $\_ \text{ ' } \rightarrow \_ :$ 
508    $(\tau_1 \tau_2 : \text{NormalType } \Delta \star) \rightarrow$ 
509    $\text{NormalType } \Delta \star$ 

```

```

510  $\forall :$ 
511    $\{\kappa : \text{Kind}\} \rightarrow (\tau : \text{NormalType } (\Delta \text{ „ } \kappa) \star) \rightarrow$ 
512    $\text{NormalType } \Delta \star$ 

```

```

513  $\mu :$ 
514    $(F : \text{NormalType } \Delta (\star \text{ ' } \rightarrow \star)) \rightarrow$ 
515    $\text{NormalType } \Delta \star$ 

```

```

516 - Qualified types

```

```

517  $\Rightarrow \_ :$ 
518    $(\pi : \text{NormalPred } \Delta \text{ R[ } \kappa_1 \text{ ]}) \rightarrow (\tau : \text{NormalType } \Delta \star) \rightarrow$ 
519    $\text{NormalType } \Delta \star$ 

```

```

520 -  $R\omega$ 

```

```

521  $\epsilon :$ 
522    $\text{NormalType } \Delta \text{ R[ } \kappa \text{ ]}$ 

```

```

523  $\triangleright \_ :$ 
524    $(l : \text{NormalType } \Delta L) \rightarrow$ 
525    $(\tau : \text{NormalType } \Delta \kappa) \rightarrow$ 
526    $\text{NormalType } \Delta \text{ R[ } \kappa \text{ ]}$ 

```

```

527 lab :
528    $(l : \text{Label}) \rightarrow$ 
529    $\text{NormalType } \Delta L$ 

```

```

530  $\lfloor \_ \rfloor :$ 
531    $(l : \text{NormalType } \Delta L) \rightarrow$ 

```

```

540   NormalType Δ ★
541 Σ :
542   (ρ : NormalType Δ R[ ★ ]) →
543   NormalType Δ ★
544
545 ΣL :
546   (ρ : NormalType Δ R[ L ]) →
547   NormalType Δ L
548

```

5.1 Renaming

We define renaming over `NormalTypes` in the same fashion as defined over `Types`. Note that we use the suffix $_k\text{NF}$ now to denote functions which operate on `NormalType` syntax. Definitions are unsurprising and omitted.

```

554 renkNE      : Renamingk Δ1 Δ2 → NeutralType Δ1 κ → NeutralType Δ2 κ
555 renkNF      : Renamingk Δ1 Δ2 → NormalType Δ1 κ → NormalType Δ2 κ
556 weakenkNF : NormalType Δ κ2 → NormalType (Δ „ κ1) κ2
557 weakenkNE : NeutralType Δ κ2 → NeutralType (Δ „ κ1) κ2
558

```

5.2 Properties of normal types

We use Agda to confirm the desired canonicity properties. First, we wish for arrow kinds to be canonically formed by λ -abstractions. This can be shown easily by induction on arrow-kinded f .

```

563 arrow-canonicity : (f : NormalType Δ (κ1 '→ κ2)) → ∃[ τ ] (f ≡ 'λ τ)
564 arrow-canonicity ('λ f) = f , refl
565

```

Second, we wish for types at row kind to be canonically either (i) a labeled type ($l \triangleright \tau$), (ii) a neutral type, or (iii) the empty row ϵ . The `row-canonicity` lemma below states precisely this. Note that we permit row-kinded types to be neutral because we do not η -expand arrow-kinded rows. Recall our discussion above that such an expansion would not be kind-preserving. This means arrow-kinded rows must be permitted to be canonically neutral.

```

571 row-canonicity : (ρ : NormalType Δ R[ κ ]) →
572   ∃[ l ] (Σ[ τ ∈ NormalType Δ κ ] ((ρ ≡ (l ▷ τ)))) or
573   Σ[ τ ∈ NeutralType Δ R[ κ ] ] (ρ ≡ ne τ) or
574   ρ ≡ ε
575
576 row-canonicity (l ▷ τ) = left (l , τ , refl)
577 row-canonicity (ne τ) = right (left (τ , refl))
578 row-canonicity ε = right (right refl)
579

```

5.3 Type embeddings

We establish an embedding back from normal types to types below. The embedding is written \Uparrow because its type is converse to our definition of normalization, written \Downarrow . We will show in later sections precisely that \Uparrow is right-inverse to \Downarrow .

```

585 ↑↑ : NormalType Δ κ → Type Δ κ
586 ↑↑NE : NeutralType Δ κ → Type Δ κ
587 ↑↑Pred : NormalPred Δ R[ κ ] → Pred Δ R[ κ ]
588

```

Much of the embedding is defined by using like-for-like constructors and recursing on the subdata.

```

 $\Uparrow\text{NE} \text{ ` } x \text{ ` } = \text{ ` } x$ 
 $\Uparrow\text{NE} (\tau_1 \cdot \tau_2) = (\Uparrow\text{NE } \tau_1) \cdot (\Uparrow\text{NE } \tau_2)$ 
 $\Uparrow\text{NE} (F \text{ <\$> } \tau) = (\Uparrow F) \text{ <\$> } (\Uparrow\text{NE } \tau)$ 
 $\Uparrow\text{Pred} (\rho_1 \cdot \rho_2 \sim \rho_3) = (\Uparrow \rho_1) \cdot (\Uparrow \rho_2) \sim (\Uparrow \rho_3)$ 
 $\Uparrow\text{Pred} (\rho_1 \lesssim \rho_2) = (\Uparrow \rho_1) \lesssim (\Uparrow \rho_2)$ 
 $\Uparrow \epsilon = \epsilon$ 
 $\Uparrow (\text{ne } x) = \Uparrow\text{NE } x$ 
 $\Uparrow (l \triangleright \tau) = (\Uparrow l) \triangleright (\Uparrow \tau)$ 
 $\Uparrow (\text{` } \lambda \tau \text{ `}) = \text{` } \lambda (\Uparrow \tau)$ 
 $\Uparrow (\tau_1 \text{ ' } \rightarrow \tau_2) = \Uparrow \tau_1 \text{ ' } \rightarrow \Uparrow \tau_2$ 
 $\Uparrow (\text{` } \forall \tau \text{ `}) = \text{` } \forall (\Uparrow \tau)$ 
 $\Uparrow (\mu \tau) = \mu (\Uparrow \tau)$ 
 $\Uparrow (\text{lab } l) = \text{lab } l$ 
 $\Uparrow \lfloor \tau \rfloor = \lfloor \Uparrow \tau \rfloor$ 
 $\Uparrow (\pi \Rightarrow \tau) = (\Uparrow\text{Pred } \pi) \Rightarrow (\Uparrow \tau)$ 

```

An exception is made for record and variant constructors, which we must reconstruct as applications:

```

 $\Uparrow (\Pi x) = \Pi \cdot \Uparrow x$ 
 $\Uparrow (\Pi L x) = \Pi \cdot \Uparrow x$ 
 $\Uparrow (\Sigma x) = \Sigma \cdot \Uparrow x$ 
 $\Uparrow (\Sigma L x) = \Sigma \cdot \Uparrow x$ 

```

6 SEMANTIC TYPES

We next define $\text{SemType } \Delta \ \kappa$, the semantic interpretation of types. SemTypes are defined by induction on the kind κ and mutually-recursively with KripkeFunctions , the interpretation of type functions.

```

 $\text{SemType} : \text{KEnv} \rightarrow \text{Kind} \rightarrow \text{Set}$ 
 $\text{KripkeFunction} : \text{KEnv} \rightarrow \text{Kind} \rightarrow \text{Kind} \rightarrow \text{Set}$ 

```

Type functions are interpreted as Kripke function spaces because they must permit arbitrary and intermediate renaming. That is, they are functions at "any world."

```

 $\text{KripkeFunction } \Delta_1 \ \kappa_1 \ \kappa_2 = (\forall \{\Delta_2\} \rightarrow \text{Renaming}_k \ \Delta_1 \ \Delta_2 \rightarrow \text{SemType } \Delta_2 \ \kappa_1 \rightarrow \text{SemType } \Delta_2 \ \kappa_2)$ 
 $\text{SemType } \Delta_1 \ (\kappa_1 \text{ ' } \rightarrow \kappa_2) = \text{KripkeFunction } \Delta_1 \ \kappa_1 \ \kappa_2$ 

```

We interpret \star and L kinded types as their normal forms.

```

 $\text{SemType } \Delta \ \star = \text{NormalType } \Delta \ \star$ 
 $\text{SemType } \Delta \ L = \text{NormalType } \Delta \ L$ 

```

We interpret rows as either nothing (the empty row), just $(\text{left } x)$ for neutral x , or just $(\text{right } (l \ , \ \tau))$ for normal l and τ . These cases correspond precisely to the three canonical forms of types with row kind.

```

638 SemType  $\Delta$  R[  $\kappa$  ] = Maybe
639   ((NeutralType  $\Delta$  R[  $\kappa$  ]) or
640    (NormalType  $\Delta$  L  $\times$  SemType  $\Delta$   $\kappa$ ))
641

```

6.1 Renaming & substitution

Renaming is defined over semantic types in an obvious fashion. Definitions are omitted except in the functional case.

```

646 renSem : Renamingk  $\Delta_1$   $\Delta_2$   $\rightarrow$  SemType  $\Delta_1$   $\kappa$   $\rightarrow$  SemType  $\Delta_2$   $\kappa$ 
647 weakenSem : SemType  $\Delta$   $\kappa_1$   $\rightarrow$  SemType ( $\Delta$  „  $\kappa_2$ )  $\kappa_1$ 
648

```

Because $R\omega\mu$ functions are interpreted into Kripke function spaces, renaming of arrow-kinded types is simply composition by the function's renaming.

```

649 renKripke : Renamingk  $\Delta_1$   $\Delta_2$   $\rightarrow$  KripkeFunction  $\Delta_1$   $\kappa_1$   $\kappa_2$   $\rightarrow$  KripkeFunction  $\Delta_2$   $\kappa_1$   $\kappa_2$ 
650 renKripke { $\Delta_1$ }  $\rho$  F { $\Delta_2$ } =  $\lambda$   $\rho'$   $\rightarrow$  F ( $\rho' \circ \rho$ )
651
652 renSem { $\kappa = \kappa' \xrightarrow{\epsilon} \kappa_1$ }  $\rho$  F = renKripke  $\rho$  F
653

```

6.2 Normalization by evaluation

Our *normalization by evaluation* proceeds in a standard fashion. We will define `reflect`, which maps neutral types to semantic types, and `reify`, which maps semantic types to normal types. We then write an evaluator that takes a Type into the semantic domain. During this process, function applications (and other forms of computation) are reduced. We finally reify the semantic type back to a normal form.

Reflection and reification are defined mutually recursively. We define the type synonym `reifyKripke` the reification of types at arrow kind, for repeated use later.

```

665 reflect :  $\forall$  { $\kappa$ }  $\rightarrow$  NeutralType  $\Delta$   $\kappa$   $\rightarrow$  SemType  $\Delta$   $\kappa$ 
666 reify :  $\forall$  { $\kappa$ }  $\rightarrow$  SemType  $\Delta$   $\kappa$   $\rightarrow$  NormalType  $\Delta$   $\kappa$ 
667 reifyKripke : KripkeFunction  $\Delta$   $\kappa_1$   $\kappa_2$   $\rightarrow$  NormalType  $\Delta$  ( $\kappa_1 \xrightarrow{\epsilon} \kappa_2$ )
668

```

Reflection of neutral types at ground kind leaves the type undisturbed.

```

671 reflect { $\kappa = \star$ }  $\tau$  = ne  $\tau$ 
672 reflect { $\kappa = L$ }  $\tau$  = ne  $\tau$ 
673 reflect { $\kappa = R[ \kappa ]$ }  $\tau$  = just (left  $\tau$ )
674

```

Reflection of neutral types at arrow kind must be η -expanded into a Kripke function. Note here that is necessary to reify the input v back to a normal type.

```

675 reflect { $\kappa = \kappa_1 \xrightarrow{\epsilon} \kappa_2$ }  $\tau = \lambda$   $\rho$   $v \rightarrow$  reflect (renkNE  $\rho$   $\tau \cdot$  reify  $v$ )
676

```

Reification similarly leaves ground types undisturbed. Semantic types at \star and label kind are already in normal form; semantic types at row kind must be translated from their semantic constructors to their NormalType constructors.

```

683 reify { $\kappa = \star$ }  $\tau = \tau$ 
684 reify { $\kappa = L$ }  $\tau = \tau$ 
685 reify { $\kappa = R[ \kappa ]$ } (just (left  $x$ )) = ne  $x$ 
686

```

```

687 reify {κ = R[ κ ]} (just (right (l , τ))) = l ▷ (reify τ)
688 reify {κ = R[ κ ]} nothing = ε
689

```

Semantic functions must be reified from Agda functions back into `NormalType` syntax. This is done by reifying the application of semantic function `F` to the reflection of the η -expanded variable λZ .

```

693 reify {κ = κ1 '→ κ2} F = reifyKripke F
694 reifyKripke {κ1 = κ1} F = 'λ (reify (F S (reflect {κ = κ1} (' Z))))
695

```

Observe that neutral types can be forced into η -long form simply by composing reification and reflection. This will prove helpful later, as the neutral type former `ne` has the same type except restricted to ground kind, but we will need to be able to promote from neutral to normal type at *all* kinds.

```

700 η-norm : NeutralType Δ κ → NormalType Δ κ
701 η-norm = reify ∘ reflect
702

```

Towards writing an evaluator, we define a semantic environment as a function mapping type variables to semantic types.

```

706 Env : KEnv → KEnv → Set
707 Env Δ1 Δ2 = ∀ {κ} → KVar Δ1 κ → SemType Δ2 κ
708

```

Environment extension and lifting can be written in a straightforward manner.

```

710 extende : (η : Env Δ1 Δ2) → (V : SemType Δ2 κ) → Env (Δ1 ,, κ) Δ2
711 lifte : Env Δ1 Δ2 → Env (Δ1 ,, κ) (Δ2 ,, κ)
712

```

The identity environment now maps type variables to semantic types. Unlike in [Chapman et al. \[2019\]](#), this environment can no longer be truly said to be an identity: type variables are de facto put into η -long form during reflection. However this change is mandatory for normalization, so we cannot define an environment that does not.

```

718 idEnv : Env Δ Δ
719 idEnv = reflect ∘ '
720

```

6.3 Helping evaluation

In aid of writing an evaluator, we found it helpful to develop *semantic* notions of the syntax introduced by $R\omega\mu$. For example, we define a type synonym for application, which is simply Agda application within the identity renaming.

```

726 _·V_ : SemType Δ (κ1 '→ κ2) → SemType Δ κ1 → SemType Δ κ2
727 F ·V V = F id V
728

```

We can further define the constructors of the three canonical forms of row-kinded types:

```

730 _▷V_ : SemType Δ L → SemType Δ κ → SemType Δ R[ κ ]
731 _▷V_ {κ = κ} ℓ τ = just (right (ℓ , τ))
732
733 ne-R : NeutralType Δ R[ κ ] → SemType Δ R[ κ ]
734 ne-R = just ∘ left
735

```

```

736  $\epsilon V : \text{SemType } \Delta R[\kappa]$ 
737
738  $\epsilon V = \text{nothing}$ 

```

The definition of semantic row mapping varies by the shape of the row V over which we are lifting. If V is neutral, so too must the mapping of F over $!V$ be neutral. Hence we reify F to normal form and leave its mapping in neutral form. If V is a labeled row ($l \triangleright \tau$), we push the application of F over τ . Finally, if V is the empty row, its mapping is empty.

```

745  $\_<\$>V\_ : \text{SemType } \Delta (\kappa_1 \xrightarrow{\quad} \kappa_2) \rightarrow \text{SemType } \Delta R[\kappa_1] \rightarrow \text{SemType } \Delta R[\kappa_2]$ 
746  $\_<\$>V\_ \{ \kappa_1 = \kappa_1 \} \{ \kappa_2 \} F (\text{just } (\text{left } x)) = \text{ne-R } (\text{reifyKripke } F \text{ } <\$> x)$ 
747  $\_<\$>V\_ \{ \kappa_1 = \kappa_1 \} \{ \kappa_2 \} F (\text{just } (\text{right } (l, \tau))) = (l \triangleright V (F \cdot V \tau))$ 
748  $\_<\$>V\_ \{ \kappa_1 = \kappa_1 \} \{ \kappa_2 \} F \text{ nothing} = \epsilon V$ 

```

Although the flap operator $_<?>V_$ is expressible as a special case of row mapping, we nevertheless find it a useful abstraction to express as a semantic function. It is defined below in terms of semantic row mapping; we find it likewise helpful to give a type synonym apply to the left hand side of this equation.

```

755  $\text{apply} : \text{SemType } \Delta \kappa_1 \rightarrow \text{SemType } \Delta ((\kappa_1 \xrightarrow{\quad} \kappa_2) \xrightarrow{\quad} \kappa_2)$ 
756  $\text{apply } a = \lambda \rho F \rightarrow F \cdot V (\text{renSem } \rho a)$ 
757
758  $\text{infixr } 0 \text{ } \_<?>V\_$ 
759  $\_<?>V\_ : \text{SemType } \Delta R[\kappa_1 \xrightarrow{\quad} \kappa_2] \rightarrow \text{SemType } \Delta \kappa_1 \rightarrow \text{SemType } \Delta R[\kappa_2]$ 
760  $f \text{ } <?>V a = \text{apply } a \text{ } <\$>V f$ 

```

Much of the latent computation in $R\omega\mu$ occurs under an outermost Π and Σ syntax. To this end, we chose to represent Π and Σ as arrow-kinded type-constants—meaning they will evaluate into Agda functions. This provides an opportunity to concisely abstract their reduction logic. We define a semantic combinator for the Π type constant below. The first two equations state that record types at \star and label kind may be formed provided normal bodies; The third equation pushes the λ -binding of F outside of the record type; the fourth equation states that application is mapping at nested row kind.

```

771  $\Pi V : \text{SemType } \Delta R[\kappa] \rightarrow \text{SemType } \Delta \kappa$ 
772  $\Pi V \{ \kappa = \star \} x = \Pi (\text{reify } x)$ 
773  $\Pi V \{ \kappa = L \} x = \Pi L (\text{reify } x)$ 
774  $\Pi V \{ \kappa = \kappa_1 \xrightarrow{\quad} \kappa_2 \} F = \lambda \rho v \rightarrow \Pi V (\text{renSem } \rho F \text{ } <?>V v)$ 
775  $\Pi V \{ \kappa = R[\kappa] \} x = (\lambda \rho v \rightarrow \Pi V v) \text{ } <\$>V x$ 

```

We can turn the semantic helper ΠV into a true Kripke function easily:

```

779  $\Pi\text{-Kripke} : \text{KripkeFunction } \Delta R[\kappa] \kappa$ 
780  $\Pi\text{-Kripke} = \lambda \rho v \rightarrow \Pi V v$ 

```

We omit the definitions of ΣV and $\Sigma\text{-Kripke}$, as they are identical modulo the use of Π constants.

6.4 Evaluation

We now write an evaluator that translates Types to semantic types; that is, translating syntactic forms to the semantic domain. A normalizer composes reification with evaluation. One can see this in the definition of `evalPred`, the predicate normalizer. (Predicates must be fully normalized as they do not have a semantic image.)

```

eval : Type  $\Delta_1 \kappa \rightarrow$  Env  $\Delta_1 \Delta_2 \rightarrow$  SemType  $\Delta_2 \kappa$ 
evalPred : Pred  $\Delta_1 R[\kappa] \rightarrow$  Env  $\Delta_1 \Delta_2 \rightarrow$  NormalPred  $\Delta_2 R[\kappa]$ 

evalPred ( $\rho_1 \cdot \rho_2 \sim \rho_3$ )  $\eta =$  reify (eval  $\rho_1 \eta$ )  $\cdot$  reify (eval  $\rho_2 \eta$ )  $\sim$  reify (eval  $\rho_3 \eta$ )
evalPred ( $\rho_1 \lesssim \rho_2$ )  $\eta =$  reify (eval  $\rho_1 \eta$ )  $\lesssim$  reify (eval  $\rho_2 \eta$ )

```

Evaluation is defined by induction over the type structure. The first three cases have types which may occur at any kind. The variable case simply uses the environment to perform a lookup; application defers to our semantic combinator `_·V_`; and evaluation of arrow types is defined recursively.

```

eval { $\kappa = \kappa$ } ( $\lambda x$ )  $\eta = \eta x$ 
eval { $\kappa = \kappa$ } ( $\tau_1 \cdot \tau_2$ )  $\eta =$  (eval  $\tau_1 \eta$ )  $\cdot V$  (eval  $\tau_2 \eta$ )
eval { $\kappa = \kappa$ } ( $\tau_1 \rightarrow \tau_2$ )  $\eta =$  (eval  $\tau_1 \eta$ )  $\rightarrow$  (eval  $\tau_2 \eta$ )

```

The next four cases are for types that only occur at kind \star . The qualified type and label singleton cases proceed by recursion over the type structure. For \forall -bound types, we must lift the environment η appropriately. In the μ case, τ has kind $\star \rightarrow \star$ and so its evaluation must be reified back to `NormalType`.

```

eval { $\kappa = \star$ } ( $\pi \Rightarrow \tau$ )  $\eta =$  evalPred  $\pi \eta \Rightarrow$  eval  $\tau \eta$ 
eval { $\kappa = \star$ } [ $\tau$ ]  $\eta =$  [eval  $\tau \eta$ ]
eval { $\kappa = \star$ } ( $\forall \tau$ )  $\eta = \forall$  (eval  $\tau$  (lifte  $\eta$ ))
eval { $\kappa = \star$ } ( $\mu \tau$ )  $\eta = \mu$  (reify (eval  $\tau \eta$ ))

```

There is only one type with exclusively label kind. Its definition is unsurprising (it houses only a `String` label).

```

eval { $\kappa = L$ } (lab  $l$ )  $\eta =$  lab  $l$ 

```

We evaluate λ -bound functions by evaluating their bodies in environments extended by the meaning their input v . Note that we are building a Kripke function and so ρ is a renaming from Δ_1 to Δ_2 and v is an input of type `SemType $\Delta_2 \kappa_1$` .

```

eval { $\kappa = \kappa_1 \rightarrow \kappa_2$ } ( $\lambda \tau$ )  $\eta = \lambda \rho v \rightarrow$  eval  $\tau$  (extende ( $\lambda \{ \kappa \} v' \rightarrow$  renSem { $\kappa = \kappa$ }  $\rho$  ( $\eta v'$ ))  $v$ )

```

Lastly, we define evaluation over the row-kinded constants and operators. As Π and Σ are represented as type constants in the Type syntax, they translate directly to the Kripke functions we defined for Π and Σ as semantic helpers. Likewise, the row mapping and labeled-row cases are interpreted immediately and desirably by their semantic helpers.

```

eval { $\kappa = R[\kappa] \rightarrow \kappa$ }  $\Pi \eta = \Pi$ -Kripke
eval { $\kappa = R[\kappa] \rightarrow \kappa$ }  $\Sigma \eta = \Sigma$ -Kripke
eval { $\kappa = R[\kappa]$ } ( $f <\$> a$ )  $\eta =$  (eval  $f \eta$ )  $<\$> V$  (eval  $a \eta$ )

```

$\text{eval } \{\kappa = _ \} (l \triangleright \tau) \eta = (\text{eval } l \eta) \triangleright_V (\text{eval } \tau \eta)$
 $\text{eval } \epsilon \eta = \epsilon_V$

Finally, we define a normalizer as the reification of evaluation.

$\Downarrow : \forall \{\Delta\} \rightarrow \text{Type } \Delta \kappa \rightarrow \text{NormalType } \Delta \kappa$
 $\Downarrow \tau = \text{reify } (\text{eval } \tau \text{ idEnv})$
 $\Downarrow_{\text{NE}} : \forall \{\Delta\} \rightarrow \text{NeutralType } \Delta \kappa \rightarrow \text{NormalType } \Delta \kappa$
 $\Downarrow_{\text{NE}} \tau = \text{reify } (\text{eval } (\Uparrow_{\text{NE}} \tau) \text{ idEnv})$

7 CORRECTNESS

We desire a normalization algorithm to remove the need for explicit type conversion proofs in terms: two types are equal iff they reduce to the same normal form, and so a normalization algorithm effectively gives a decision procedure for type equivalence. It next falls upon us to verify that this normalization algorithm indeed respects our syntactic account of type equivalence. How we do so is fairly routine to other normalization-by-evaluation efforts. We first show that the algorithm is complete with respect to syntactic type equivalence:

completeness : $\forall \{\tau_1 \tau_2 : \text{Type } \Delta \kappa\} \rightarrow \tau_1 \equiv \tau_2 \rightarrow \Downarrow \tau_1 \equiv \Downarrow \tau_2$

Completeness here states that equivalent types normalize to the same types. Conversely, soundness states that every type is equivalent to its normalization. (In particular, every type is equivalent to the normalization of its embedding.)

soundness : $\forall \{\Delta_1 \kappa\} \rightarrow (\tau : \text{Type } \Delta_1 \kappa) \rightarrow \tau \equiv \Uparrow (\Downarrow \tau)$

7.1 A logical relation

We will prove completeness using a logical relation on semantic types. We would like to be able to equate semantic types, but they prove to be "too large": in particular, our definition of Kripke functions permit functions which may not respect composition of renaming. The solution is to reason about semantic types modulo a partial equivalence relation (PER) that both respects renamings (which we call *uniformity*) and also equates functions extensionally. We write $\tau_1 \approx \tau_2$ to denote that the semantic types τ_1 and τ_2 are equivalent modulo this relation. For clarity, we give names to the two properties (*uniformity* and *point equality*) we desire related types to hold, and define them mutually recursively.

$_ \approx _ : \text{SemType } \Delta \kappa \rightarrow \text{SemType } \Delta \kappa \rightarrow \text{Set}$
 $\text{PointEqual} \approx : \forall \{\Delta_1\} \{\kappa_1\} \{\kappa_2\} (F G : \text{KripkeFunction } \Delta_1 \kappa_1 \kappa_2) \rightarrow \text{Set}$
 $\text{Uniform} : \forall \{\Delta\} \{\kappa_1\} \{\kappa_2\} \rightarrow \text{KripkeFunction } \Delta \kappa_1 \kappa_2 \rightarrow \text{Set}$

We define $_ \approx _$ recursively over the kind of its equated types. In the first two cases, τ_1 and τ_2 are normal types, which we equate propositionally. In the third case, we assert that Kripke functions F and G are uniform and point-equal to one another. Uniformity asserts a certain commutativity of renaming: you may either rename the result of applying F to V_1 , or you may rename F before applying it to a renamed input. point equality on Kripke functions F and G asserts that F and G take related inputs to related outputs. The latter property is what one should expect of a logical relation; the former property can be attributed to [Chapman et al. \[2019\]](#), who in turn attribute [Allais et al. \[2013\]](#).

883 **Uniform** $\{\Delta_1\} \{\kappa_1\} \{\kappa_2\} F =$
 884 $\forall \{\Delta_2 \Delta_3\} (\rho_1 : \text{Renaming}_k \Delta_1 \Delta_2) (\rho_2 : \text{Renaming}_k \Delta_2 \Delta_3) (V_1 V_2 : \text{SemType } \Delta_2 \kappa_1) \rightarrow$
 885 $V_1 \approx V_2 \rightarrow (\text{renSem } \rho_2 (F \rho_1 V_1)) \approx (\text{renKripke } \rho_1 F \rho_2 (\text{renSem } \rho_2 V_2))$
 886
 887 **PointEqual- \approx** $\{\Delta_1\} \{\kappa_1\} \{\kappa_2\} F G =$
 888 $\forall \{\Delta_2\} (\rho : \text{Renaming}_k \Delta_1 \Delta_2) \{V_1 V_2 : \text{SemType } \Delta_2 \kappa_1\} \rightarrow$
 889 $V_1 \approx V_2 \rightarrow F \rho V_1 \approx G \rho V_2$
 890
 891 $_ \approx _ \{ \kappa = \star \} \tau_1 \tau_2 = \tau_1 \equiv \tau_2$
 892 $_ \approx _ \{ \kappa = \mathbf{L} \} \tau_1 \tau_2 = \tau_1 \equiv \tau_2$
 893 $_ \approx _ \{ \Delta_1 \} \{ \kappa = \kappa_1 \xrightarrow{\text{green}} \kappa_2 \} F G =$
 894 **Uniform** $F \times \text{Uniform } G \times \text{PointEqual-}\approx \{ \Delta_1 \} F G$
 895

896 The last six cases are over row kinded semantic types. The first case states that neutral rows
 897 must be propositionally equal; the second states that two rows of the form $(\mathbf{l}_1 \triangleright \tau_1)$ and $(\mathbf{l}_2 \triangleright \tau_2)$
 898 are related iff their labels are equal and their types are related. The third case states that the empty
 899 row is related to itself (which is always true). All other cases are nonsensical, and so are set to \perp .

900 $_ \approx _ \{ \kappa = \mathbf{R} [\kappa] \} (\text{just } (\text{left } x)) (\text{just } (\text{left } y)) = x \equiv y$
 901 $_ \approx _ \{ \kappa = \mathbf{R} [\kappa] \} (\text{just } (\text{right } (l_1, \tau_1))) (\text{just } (\text{right } (l_2, \tau_2))) = l_1 \equiv l_2 \times \tau_1 \approx \tau_2$
 902 $_ \approx _ \{ \kappa = \mathbf{R} [\kappa] \} \text{nothing nothing} = \top$
 903 $_ \approx _ \{ \kappa = \mathbf{R} [\kappa] \} (\text{just } _) (\text{just } _) = \perp$
 904 $_ \approx _ \{ \kappa = \mathbf{R} [\kappa] \} (\text{just } _) \text{nothing} = \perp$
 905 $_ \approx _ \{ \kappa = \mathbf{R} [\kappa] \} \text{nothing } (\text{just } _) = \perp$
 906

907
 908 **7.1.1 Properties of the completeness relation.** The completeness relation forms a *partial equivalence*
 909 *relation* (PER). As uniformity is a unary property, it follows quickly that $_ \approx _$ cannot be reflexive,
 910 but a limited form of reflexivity does hold: provided that V is related to *some* other V' , it relates to
 911 itself. The other properties (symmetry and transitivity) are simple enough to show. We introduce
 912 two helpers, $\text{refl-}\approx_l$ and $\text{refl-}\approx_r$ to describe left and right reflexive projections.

913 $\text{refl-}\approx_l : \forall \{V_1 V_2 : \text{SemType } \Delta \kappa\} \rightarrow V_1 \approx V_2 \rightarrow V_1 \approx V_1$
 914 $\text{refl-}\approx_r : \forall \{V_1 V_2 : \text{SemType } \Delta \kappa\} \rightarrow V_1 \approx V_2 \rightarrow V_2 \approx V_2$
 915 $\text{sym-}\approx : \forall \{\tau_1 \tau_2 : \text{SemType } \Delta \kappa\} \rightarrow \tau_1 \approx \tau_2 \rightarrow \tau_2 \approx \tau_1$
 916 $\text{trans-}\approx : \forall \{\tau_1 \tau_2 \tau_3 : \text{SemType } \Delta \kappa\} \rightarrow \tau_1 \approx \tau_2 \rightarrow \tau_2 \approx \tau_3 \rightarrow \tau_1 \approx \tau_3$
 917

918
 919 we commonly invoke two main lemmas. $\text{reflect-}\approx$ reflects propositional equality to semantic
 920 equivalence, and $\text{reify-}\approx$ reifies equivalent semantic types to propositional equality. We make
 921 great use of the latter lemma, which states intuitively that related types should have the same
 922 reifications. One may alternatively think of this lemma as congruence of reification modulo semantic
 923 equivalence.

924 $\text{reflect-}\approx : \forall \{\tau_1 \tau_2 : \text{NeutralType } \Delta \kappa\} \rightarrow \tau_1 \equiv \tau_2 \rightarrow \text{reflect } \tau_1 \approx \text{reflect } \tau_2$
 925 $\text{reify-}\approx : \forall \{\tau_1 \tau_2 : \text{SemType } \Delta \kappa\} \rightarrow \tau_1 \approx \tau_2 \rightarrow \text{reify } \tau_1 \equiv \text{reify } \tau_2$
 926

7.2 The fundamental theorem & completeness

927
 928 We would like to show that all well-kinded equivalent types have semantically equivalent evalua-
 929 tions. Completeness follows shortly thereafter. The fundamental theorem for completeness (fundC)
 930

states that equivalent types evaluate to related types under related environments. Towards this goal, we first define a point-wise equivalence on semantic environments.

$\text{Env} \approx : (\eta_1 \ \eta_2 : \text{Env } \Delta_1 \ \Delta_2) \rightarrow \text{Set}$
 $\text{Env} \approx \eta_1 \ \eta_2 = \forall \{ \kappa \} (x : \text{KVar } _ \ \kappa) \rightarrow (\eta_1 \ x) \approx (\eta_2 \ x)$

We show that related environments remain related when extended with related arguments.

$\text{extend} \approx : \forall \{ \eta_1 \ \eta_2 : \text{Env } \Delta_1 \ \Delta_2 \} \rightarrow \text{Env} \approx \eta_1 \ \eta_2 \rightarrow$
 $\{ V_1 \ V_2 : \text{SemType } \Delta_2 \ \kappa \} \rightarrow$
 $V_1 \approx V_2 \rightarrow$
 $\text{Env} \approx (\text{extende } \eta_1 \ V_1) (\text{extende } \eta_2 \ V_2)$

It is easy to show as well that the identity environment relates to itself.

$\text{idEnv} \approx : \forall \{ \Delta \} \rightarrow \text{Env} \approx (\text{idEnv } \{ \Delta \}) (\text{idEnv } \{ \Delta \})$
 $\text{idEnv} \approx x = \text{reflect} \approx \text{refl}$

We may now state the fundamental theorem for completeness. Again, as we have no semantic image of predicates, the fundamental theorem for predicates simply asserts that the evaluation of equivalent predicates are propositional equal.

$\text{fundC} : \forall \{ \tau_1 \ \tau_2 : \text{Type } \Delta_1 \ \kappa \} \{ \eta_1 \ \eta_2 : \text{Env } \Delta_1 \ \Delta_2 \} \rightarrow$
 $\text{Env} \approx \eta_1 \ \eta_2 \rightarrow \tau_1 \equiv \tau_2 \rightarrow \text{eval } \tau_1 \ \eta_1 \approx \text{eval } \tau_2 \ \eta_2$
 $\text{fundC-pred} : \forall \{ \pi_1 \ \pi_2 : \text{Pred } \Delta_1 \ \text{R}[\kappa] \} \{ \eta_1 \ \eta_2 : \text{Env } \Delta_1 \ \Delta_2 \} \rightarrow$
 $\text{Env} \approx \eta_1 \ \eta_2 \rightarrow \pi_1 \equiv \pi_2 \rightarrow \text{evalPred } \pi_1 \ \eta_1 \equiv \text{evalPred } \pi_2 \ \eta_2$

Completeness follows immediately as a special case of the fundamental theorem.

$\text{Completeness} : \forall \{ \tau_1 \ \tau_2 : \text{Type } \Delta \ \kappa \} \rightarrow \tau_1 \equiv \tau_2 \rightarrow \Downarrow \tau_1 \equiv \Downarrow \tau_2$
 $\text{Completeness } eq = \text{reify} \approx (\text{fundC } \text{idEnv} \approx eq)$

8 SOUNDNESS

Soundness states that every type is equivalent to its normalization. Intuitively, completeness tells us that all "computation" inherent in the equivalence relation is captured by normalization; coversely, soundness tells us that all computation inherent in the normalization algorithm is declared in the equivalence relation.

8.1 A logical relation

We prove soundness by a separate logical relation that relates (unnormalized) types to semantic types. We write $\llbracket \tau \rrbracket \approx V$ to denote that the type τ is related to the semantic type V . This syntax is inspired by the result we wish to show: that evaluating τ yields a semantic type V . We give the type synonym `SoundKripke` for the functional case.

$\text{infix } 0 \llbracket _ \rrbracket \approx _$
 $\llbracket _ \rrbracket \approx _ : \forall \{ \kappa \} \rightarrow \text{Type } \Delta \ \kappa \rightarrow \text{SemType } \Delta \ \kappa \rightarrow \text{Set}$
 $\text{SoundKripke} : \text{Type } \Delta_1 \ (\kappa_1 \xrightarrow{\text{c}} \kappa_2) \rightarrow \text{KripkeFunction } \Delta_1 \ \kappa_1 \ \kappa_2 \rightarrow \text{Set}$

In two of the ground cases, V is a normal type, and so we simply assert type equivalence with the normal type's embedding.

$$\llbracket _ \rrbracket \approx _ \{ \kappa = \star \} \tau \ V = \tau \equiv t \uparrow V$$

$$\llbracket _ \rrbracket \approx _ \{ \kappa = L \} \tau \ V = \tau \equiv t \uparrow V$$

In the row case, we assert (resp.) that (i) if τ relates to nothing then it must be equivalent to the empty row; (ii) if τ relates to a neutral row then it must be equivalent to a neutral row; and (iii) if τ relates to a labeled row then it must be equivalent to a labeled row and that labeled row's component type must relate to itself.

$$\llbracket _ \rrbracket \approx _ \{ \kappa = R[\kappa] \} \tau \ \text{nothing} = \tau \equiv t \ \epsilon$$

$$\llbracket _ \rrbracket \approx _ \{ \kappa = R[\kappa] \} \tau \ (\text{just} \ (\text{left} \ n)) = \tau \equiv t \ (\uparrow \text{NE} \ n)$$

$$\llbracket _ \rrbracket \approx _ \{ \kappa = R[\kappa] \} \tau \ (\text{just} \ (\text{right} \ (l, v))) = (\tau \equiv t \uparrow (l \triangleright \text{reify} \ v)) \times (\llbracket \uparrow (\text{reify} \ v) \rrbracket \approx v)$$

In the functional case, we assert that logically related functions map related inputs to related outputs.

$$\llbracket _ \rrbracket \approx _ \{ \Delta_1 \} \{ \kappa = \kappa_1 \xrightarrow{\epsilon} \kappa_2 \} f \ F = \text{SoundKripke} \ f \ F$$

$$\text{SoundKripke} \ \{ \Delta_1 = \Delta_1 \} \{ \kappa_1 = \kappa_1 \} \{ \kappa_2 = \kappa_2 \} f \ F =$$

$$(\forall \{ \Delta_2 \} (\rho : \text{Renaming}_k \ \Delta_1 \ \Delta_2) \{ v \ V \} \rightarrow$$

$$\llbracket v \rrbracket \approx V \rightarrow$$

$$\llbracket (\text{ren}_k \ \rho \ f \cdot v) \rrbracket \approx (\text{renKripke} \ \rho \ F \cdot V))$$

8.2 Properties of the soundness relation

We reflect type equivalence to the relation and reify the relation to type equivalence as so.

$$\text{reflect-}\llbracket _ \rrbracket \approx : \forall \{ \tau : \text{Type} \ \Delta \ \kappa \} \{ v : \text{NeutralType} \ \Delta \ \kappa \} \rightarrow$$

$$\tau \equiv t \uparrow \text{NE} \ v \rightarrow \llbracket \tau \rrbracket \approx (\text{reflect} \ v)$$

$$\text{reify-}\llbracket _ \rrbracket \approx : \forall \{ \tau : \text{Type} \ \Delta \ \kappa \} \{ V : \text{SemType} \ \Delta \ \kappa \} \rightarrow$$

$$\llbracket \tau \rrbracket \approx V \rightarrow \tau \equiv t \uparrow (\text{reify} \ V)$$

8.3 The fundamental theorem & soundness

Towards defining the fundamental theorem, we first define a relation between syntactic environments (substitutions) and semantic environments. Intuitively, the substitution σ is related to the environment η if each type mapped to by σ point-wise relates to the semantic type mapped to by η .

$$\llbracket _ \rrbracket \approx e_ : \forall \{ \Delta_1 \ \Delta_2 \} \rightarrow \text{Substitution}_k \ \Delta_1 \ \Delta_2 \rightarrow \text{Env} \ \Delta_1 \ \Delta_2 \rightarrow \text{Set}$$

$$\llbracket _ \rrbracket \approx e_ \{ \Delta_1 \} \sigma \ \eta = \forall \{ \kappa \} (\alpha : \text{KVar} \ \Delta_1 \ \kappa) \rightarrow \llbracket (\sigma \ \alpha) \rrbracket \approx (\eta \ \alpha)$$

The fundamental theorem for soundness states that the substitution of τ by σ is related to the evaluation of τ by η . Intuitively, substitution may be thought of as a syntactic notion of evaluation, and hence we are stating that syntactic and semantic evaluations relate.

$$\text{fundS} : \forall \{ \Delta_1 \ \Delta_2 \ \kappa \} (\tau : \text{Type} \ \Delta_1 \ \kappa) \{ \sigma : \text{Substitution}_k \ \Delta_1 \ \Delta_2 \} \{ \eta : \text{Env} \ \Delta_1 \ \Delta_2 \} \rightarrow$$

$$\llbracket \sigma \rrbracket \approx e \ \eta \rightarrow \llbracket \text{sub}_k \ \sigma \ \tau \rrbracket \approx (\text{eval} \ \tau \ \eta)$$

We show that the identity substitution id is related to the identity environment:

$$\text{idSR} : \forall \{ \Delta_1 \} \rightarrow \llbracket _ \rrbracket \approx e (\text{idEnv} \ \{ \Delta_1 \})$$

$$\text{idSR} \ \alpha = \text{reflect-}\llbracket _ \rrbracket \approx \text{eq-refl}$$

and also show that \sim is indeed an identity substitution: it fixes the meaning of the type over which it is substituted.

$\text{sub}_k\text{-id} : \forall (\tau : \text{Type } \Delta \kappa) \rightarrow \text{sub}_k \tau \equiv \tau$

Soundness follows as a special case of the fundamental theorem.

$\text{Soundness} : \forall \{\Delta_1 \kappa\} \rightarrow (\tau : \text{Type } \Delta_1 \kappa) \rightarrow \tau \equiv \uparrow (\Downarrow \tau)$

$\text{Soundness } \tau = \text{subst } (_ \equiv \uparrow (\Downarrow \tau)) (\text{sub}_k\text{-id } \tau) ((\text{reify-}\llbracket _ \rrbracket \approx (\text{fundS } \tau \text{ idSR})))$

9 STABILITY

Stability states that embedding \uparrow is a right-inverse to normalization \Downarrow , or, in categorical terms, that \Downarrow is a split-monomorphism.

$\text{stability} : \forall (\tau : \text{NormalType } \Delta \kappa) \rightarrow \Downarrow (\uparrow \tau) \equiv \tau$

It is desirable that a normalization algorithm adheres to this property, as it states effectively that there is "no more work" to be done by re-normalization. Indeed, both idempotency and surjectivity are implied.

$\text{idempotency} : \forall (\tau : \text{Type } \Delta \kappa) \rightarrow (\uparrow (\Downarrow (\uparrow (\Downarrow \tau)))) \equiv \uparrow (\Downarrow \tau)$

$\text{idempotency } \tau \text{ rewrite stability } (\Downarrow \tau) = \text{refl}$

$\text{surjectivity} : \forall (\tau : \text{NormalType } \Delta \kappa) \rightarrow \exists [v] (\Downarrow v \equiv \tau)$

$\text{surjectivity } \tau = (\uparrow \tau, \text{stability } \tau)$

10 REMARK

10.1 Comparison to Chapman et al. [2019]

Our mechanization has closely resembled that of Chapman et al. [2019]. Our definition of semantic types, however, has differed, as our normalization is with respect to both β - and η -equivalence, whereas Chapman et al's is simply β -equivalence. Changing this definition simplifies some things and complicates others. The definition of semantic types is simpler: whereas Chapman et al permit function types to be interpreted as NeutralTypes, ours must be interpreted into solely Kripke function spaces. This complicates the definitions of reify and reflect, which must become mutually recursive, as we are unable to reflect neutral types at arrow kind to neutral types. We will show later that some of Chapman et al's metatheory relies on neutral forms to not be disturbed by normalization. This complicates the definition of term-level, normality-preserving substitution.

REFERENCES

- Guillaume Allais, Pierre Boutillier, and Conor McBride. New equations for neutral terms: A sound and complete decision procedure, formalized, 2013. URL <https://arxiv.org/abs/1304.0809>.
- James Chapman, Roman Kireev, Chad Nester, and Philip Wadler. System F in agda, for fun and profit. In Graham Hutton, editor, *Mathematics of Program Construction - 13th International Conference, MPC 2019, Porto, Portugal, October 7-9, 2019, Proceedings*, volume 11825 of *Lecture Notes in Computer Science*, pages 255–297. Springer, 2019. ISBN 978-3-030-33635-6. doi: 10.1007/978-3-030-33636-3_10. URL https://doi.org/10.1007/978-3-030-33636-3_10.
- Alex Hubers and J. Garrett Morris. Generic programming with extensible data types: Or, making ad hoc extensible data types less ad hoc. *Proc. ACM Program. Lang.*, 7(ICFP):356–384, 2023. doi: 10.1145/3607843. URL <https://doi.org/10.1145/3607843>.
- Philip Wadler, Wen Kokke, and Jeremy G. Siek. *Programming Language Foundations in Agda*. August 2022. URL <https://plfa.inf.ed.ac.uk/20.08/>.